

ESL Reference Guide

Tony Clark

Aston University, Birmingham, UK

Abstract

ESL is an actor-based functional language that aims to provide abstractions over applications that require a high degree of concurrency. ESL is statically typed which means that programs can be verified for type safety before they run. ESL provides extensions over the standard Actor Model, including shared state and access to the underlying Java platform, for practical application development. ESL compiles to pure Java and runs against a small Java library. The ESL compiler is written in ESL. ESL is supported by a development environment called EDB that includes libraries for data storage using Excel, and data visualisation in terms of HTML, SVG and GraphViz. This report provides a reference for ESL and EDB using examples.

Contents

1	Introduction	42
2	Data and Types	42
3	Blocks and Commands	42
4	List Comprehensions and List Operations	44
5	Algebraic Data Types and Pattern Matching	44
6	Sets and Collection Matching	44
7	Actors	5
8	Binding and Scope	5
9	Modules	7
10	Time	8
11	State	8
12	Concurrency	9
12.1	Search	9
12.2	Quicksort	10
12.3	Termites	10
12.4	Dining Philosophers	11
12.5	Segregation	13
12.6	Services	15
13	Polymorphism	15
13.1	Map/Reduce	17
13.1.1	All Words of a Given Length	18
13.1.2	Occurrences of a Word	18
13.2	Cached Functions	18
14	Object Orientation	19
15	Observers	22
15.1	A Taxi Service	24
16	EDB	26
16.1	EDB Interface	26
16.2	EDB Actor	27
16.3	EDB Displays	28
16.4	Tables	28
16.5	Pie Charts	28
16.6	Line Graphs	29
16.6.1	Supply Chains	29
16.7	Pictures	31
16.8	Graphs	33
16.9	Filmstrips	35
16.10	Sequence Diagrams	36
16.11	Combining Pictures	37
17	Excel Interface	40
18	Java Interface	41
19	Compilation	42
20	Implementation	42
21	Bibliography	42
A	ESL Syntax	44
B	Displays	44
C	List Operations	44

1. Introduction

This document describes the language ESL and its implementation. ESL is an actor-based language that has been designed to offer a convenient way to build applications that benefit from a large number of independent concurrent processes that communicate in terms of structured data. ESL compiles directly to Java source code and can therefore easily integrate with Java applications. The benefit of using ESL compared to Java is the abstraction that it provides over the construction and management of processes and the data that is passed between them. In addition, ESL enforces a separation of concerns since communication is by asynchronous message which reduces the risks associated with large scale concurrency due to race conditions and shared data. Having said that, whilst ESL actors are encapsulated, it is unreasonable to build a large scale system without some degree of shared data, so ESL provides mechanisms for data sharing where needed. ESL aims to support system verification through an expressive statically checked type system.

The Actor Model of Computation is characterised in figure 1 which shows three actors. Each actor has a local thread of control which selects a message from the head of the mailbox and processes it. When the current message is processed, the thread inspects the mailbox for the next message; if no message is available then the actor becomes idle until a message is received. Each actor has a local state that can be inspected and updated by the methods that handle the messages.

An actor can send messages to any actor that it knows. An actor will know about an actor if it created the actor or it re-




Figure 1: Actor Model of Computation [24]

ceived the actor as part of a message. A message may contain basic data items, collections of data item, actors, and functions. When a source actor passes a function to a target actor, the source context is captured by the function. When the target actor calls such a function, it runs on the target thread whilst accessing the source context that was captured when the function was created.

2. Data and Types

ESL provides basic data types: integers, floats, booleans and strings. For example the following listing declares identifiers along with their values. Comments follow the Java format:

```

1 /* The parameters of the program are defined below.
2 Change the starting position to get a different
3 configuration. */
4
5 xx:Int      = 100;      // Starting x position.
6 yy:Int      = 200;      // Starting y position.
7 pi:Float    = 3.14;     // Used for position calculation.
8 isActive:Bool = true;   // True when moving.
9 name:Str    = 'Wilma'; // Used as a unique identifier.

```

Characters such as #a and #+ are represented as integers ascii codes and are therefore of type `Int`. Special characters are introduced using #\ followed by their name: #\space, #\newline and #\backslash.

ESL supports homogeneous arrays, for example the following is a one-dimensional array of strings with 100 elements indexed from 0:

```

1 names:Array[Str] = new Array[Str](100);
2 names[0] := 'Wilma';
3 print[Str]('value at 0 = ' + names[0]);

```

Arrays may be nested, for example to create a two-dimensional array of booleans:

```

1 points:Array[Array[Bool]] = new Array[Array[Bool]](100);
2 for x:Int in 0..100 do {
3   points[x] := new Array[Bool](200);
4   for y:Int in 0..200 do {

```

```

5     points[x][y] := true;
6   }
7 }

```

Note that array elements can be updated and passed between actors, therefore they are a basis for race conditions and other concurrency problems. The intention is that arrays are used locally within an actor as efficient indexable storage. If arrays are shared then access should be protected using locks.

ESL provides homogeneous lists whose elements cannot be modified by side effect and are therefore safe to pass between actors. A list is constructed using : and [...] and can be decomposed using the operations `head` and `tail` in addition to pattern matching (see below). For example:

```

1 nums:[Int] = [1,2,3,4,5,6,7,8,9,0];
2 print[Int](head[Int](nums));
3 print[[Int]](tail[Int](nums))
4 print[[[Int]]](head[Int](nums) : tail[Int](tail[Int](nums)));

```

which produces 1, [2,3,4,5,6,7,8,9,0] and [1,3,4,5,6,7,8,9,0] respectively. Note that the operations `head` and `tail` are supplied with the type of the list element; this is because these operations are *polymorphic* as described in section 13. The definition of `nums` above is equivalent to:

```
1 nums:[Int] = 1:2:3:4:5:6:7:8:9:0:[];
```

Sets and bags are like lists except the elements are not ordered. Adding an element to a set or bag produces a new set or bag respectively. Adding an element to a set has no effect if the set already contains the element, otherwise both the set and bag are extended:

```

1 setOfNames:Set[Str] = set{'Fred','Wilma'};
2 bagOfNames:Bag[Str] = bag{'Fred','Wilma'};
3 print[Set[Str]](set{'Fred' | setOfNames});
4 print[Bag[Str]](bag{'Fred' | setOfNames});

```

produces `set{'Fred','Wilma'}` and `bag{'Fred','Wilma','Fred'}` respectively. Note that extraction from sets and bags occurs via pattern matching as described below. Sets require a definition of equality for ESL values which is provided in figure 2.

ESL hash tables map keys to values. A hash table provides operations for adding a mapping, accessing the value for a key, testing whether a mapping exists. A hash table uses field reference to get the list of keys and the list of values:

```

1 ages:Hash[Str,Int] = new Hash[Str,Int];
2 ages.put('Fred',35);
3 ages.put('Wilma',32);
4 print[Bool](ages.hasKey('Fred'));
5 print[Int](ages.get('Wilma'));
6 print[[Str]](ages.keys);
7 print[[Int]](ages.vals);

```

prints `true`, 32, `[Fred,Wilma]`, and `[35,32]`. Note that because hash tables are updated by side effect, they are not thread safe and should be used locally within an actor, or access should be protected using locks.

ESL functions are first-class values and may be passed between actors within data structures. The rules for identifier binding within functions are defined in section 8. A simple

Type	Equality for Values of Type
i1,i2:Int	i1 and i2 are the same number
f1,f2:Float	f1 and f2 are the same number
s1,s2:Str	s1 and s2 have the same characters in order
b1,b2:Bool	b1 and b2 are the same boolean values
h1:11,h2:12:[T]	h1 is equal to h2, and 11 is equal to 12
[]:[T]	equal to an empty list of the same type.
set{e1::T s1::Set{T}} set{e2::T s2::Set{T}}	e1 and e2 are equal, and s1 and s2 are equal
set{}::Set{T}	is only equal to an empty set of the same type.
bag{e1::T b1:Bag{T}} bag{e2::T b2:Bag{T}}	e1 and e2 are equal, and b1 and b2 are equal
bag{}::Bag{T}	is only equal to an empty bag of the same type.
a1,a2:Array[T]	a1 and a2 are the same array
t1,t2:Hash[K,V]	t1 and t2 are the same table
a1,a2:Act{...}	a1 and a2 are the same actor
f1,f2:(T,...) → T	f1 and f2 are the same function
r1,r2:{n→T,...}	r1.n is equal to r2.n for all names
C(n1,...), C(m2,...)	equal when corresponding ni and mi are equal
null[T]	only equal to the value null

Figure 2: Data Equality

example of a function is:

```
1 add1:(Int)→Int = fun(x:Int)::Int x + x;
2 print[Int](add1(100));
```

which prints 101.

Records map names to values. Unlike hash tables, the domain of a record is fixed. New type names can be defined that refer to existing types, so the following code implements an two-dimensional point object:

```
1 type Point = { x:Int; y:Int }
2 mkPoint(n:Int,m:Int)::Point = {x→n; y→m};
3 print[Point](mkPoint(100,200));
4 print[Int](mkPoint(100,200).x);
```

prints {x=100,y=200} and 100. Records cannot be updated by side effect.

The value `null` can be used as an undefined value for any data type. It is polymorphic and is followed by the type for which it acts as undefined. For example:

```
1 x:Int = null[Int];
2 p:Point = null[Point];
```

The following operations are builtin to ESL:

```
1 - + - :: (Int,Int) → Int // addition
2 - + - :: (Float,Float) → Float // addition
3 - + - :: (Str,T) → Str // concatenation
4 - + - :: (T,Str) → Str // concatenation
5 - + - :: ([T],[T]) → [T] // concatenation
6 - + - :: (Set{T},Set{T}) → Set{T} // union
7 - - - :: (Int,Int) → Int // subtraction
8 - - - :: (Float,Float) → Float // subtraction
9 - * - :: (Int,Int) → Int // multiplication
10 - * - :: (Float,Float) → Float // multiplication
11 - / - :: (Int,Int) → Int // division
12 - / - :: (Float,Float) → Float // division
13 - % - :: (Int,Int) → Int // modulo
14 - = - :: (T,T) → Bool // equal to
15 - <> - :: (T,T) → Bool // not equal to
```

16 _ and _ :: (Bool,Bool) → Bool	// boolean and
17 _ or _ :: (Bool,Bool) → Bool	// boolean or
18 not :: (Bool) → Bool	// boolean not
19 _ .. _ :: (Int,Int) → [Int]	// integer range
20 _ @ _ :: (T,T) → T	// parallel
21 print[T] :: (T) → Void	// print
22 random :: (Int) → Int	// random value
23 wait :: (Int) → Void	// thread delay
24 isqrt :: (Int). → Float	// square root
25 intToFloat :: (Int) → Float	// conversion
26 round :: (Float) → Int	// conversion
27 stopAll :: () → Void	// halt execution

When an ESL identifier is defined its type must be declared. Types are defined as follows:

Type	Description
Int	integers
Float	floats
Bool	booleans
Str	strings
Void	nothing
[T]	lists
(T,...) → T	functions
Array[T]	arrays
Set{T}	sets
Bag{T}	bags
Hash[T,T]	hash tables
{i:T;...}	records
Forall[V,...] T	polymorphic values
rec V. T	recursive types
union { C(T,...);... }	union types

A type may be named and then the name can be used as a shorthand. For example:

```
1 type Ints = [Int];
2 type AddInts = (Ints,Ints) → Ints;
3
4 addInts:AddInts = fun(l1:Ints,l2:Ints)::Ints
5 case l1,l2 {
6   [],l → [];
7   l,[] → [];
8 }
```

```

8     x:11,y:12 → (x+y):addInts(11,12);
9 }
```

A function binding can be defined in-line so that the example above can be expressed as follows:

```

1 addInts(11:Ints,12:Ints)::Ints =
2   case 11,12 {
3     [] ,1 → [];
4     1,[] → [];
5     x:11,y:12 → (x+y):addInts(11,12);
6   }
```

Types can be recursive, for example the following implements two dimensional points with function updates:

```

1 type Point2D = {
2   getX():Int;
3   getY():Int;
4   setX:(Int)→Point2D;
5   setY:(Int)→Point2D
6 };
7
8 mkPoint(x:Int,y:Int)::Point2D = {
9   getX → fun()::Int x;
10  getY → fun()::Int y;
11  setX → fun(x:Int)::Point2D mkPoint(x,y);
12  setY → fun(y:Int)::Point2D mkPoint(x,y)
13 };
14
15 let p:Point2D = mkPoint(100,200);
16 in {
17   print[Int](p.getX());
18   print[Int](p.setX(p.getX()+1).getX());
19 }
```

which prints 100 then 101.

3. Blocks and Commands

The type `Void`, like Java, represents an evaluation that does not return a result. The operation `print[T](t)` takes a value `t::T` and ‘returns’ `Void`. A *block* is a sequence of *commands* enclosed in curly brackets. A command is just any expression, but syntactically, if the expression does not end with a closing `}`, then the command must be terminated with a semicolon. The return value of a block is the return value of the last command. For example:

```

1 print3(a:Int,b:Int,c:Int)::Void = {
2   print[Int]('first = ' + a);
3   print[Int]('second = ' + b);
4   print[Int]('third = ' + b);
5 }
```

The special case of the block `{}` is of type `Void`.

4. List Comprehensions and List Operations

Lists are a very useful way of organising collections of elements. They can easily be processed using recursion because of their structure: a list is either empty `[]` or is a list `1` with a head element `e` ‘consed’ to the tail: `e:1`.

A *list comprehension* is an expression that transforms lists. It has the form: `[e | q ...]` where `e` is an expression and `q`

is a *qualifier*. The idea is that a list is constructed whose elements are formed by evaluating `e` after performing each of the qualifiers in turn. There are two types of qualifier: *binding* and *predicate*. A binding qualifier has the form `p ← e` where `p` is a pattern and `e` is an expression whose value is a list. When it is evaluated, a binding qualifier repeatedly matches the elements from the list against the pattern. A predicate qualifier has the form `?e` where `e` is a boolean valued expression. When it is evaluated, if the expression in a predicate qualifier produces `true` then execution continues, otherwise it ignores the element most recently selected by a binding qualifier and selects the next element. For example:

```
1 [ n | n ← 0..100, ?(n % 2 = 0) ]
```

produces a list of even numbers between 0 and 99 inclusive. Multiple binding qualifiers may be used:

```
1 [ mkPoint(x,y) | x ← 0..100, y ← 0..200, ?(x < y) ]
```

Lists can be processed using recursive functions and pattern matching. Typically a case-expression is used to match against a base case (often `[]`) and a recursion case (often `h:t`). For example the following operation adds up the distances of all points to the origin:

```

1 addDists(ps:[Point]):Float =
2   case ps {
3     [] → 0.0;
4     h:t → distanceToOrigin(h) + addDists(t);
5   }
```

The following defines Quicksort in ESL:

```

1 quicksort(l:[Int]):[Int] =
2   case l {
3     [] → [];
4     x:1 →
5       quicksort([ n | n ← 1, ?(n < x) ])
6       + [x] +
7       quicksort([ n | n ← 1, ?(n > x) ]);
8   }
```

The collection of list operations that are supported by ESL are shown in C.

5. Algebraic Data Types and Pattern Matching

ESL supports algebraic data types, also known as *union types*. For example, the directions of the compass can be represented using different integers or strings, but it is better to define a new type `Direction` and to have 4 unique values of that type. This can be defined as follows:

```

1 data Direction =
2   North
3   | South
4   | East
5   | West;
```

Now any identifier defined to have type `Direction` can hold one of the 4 values and the special value `null[Direction]`:

```
1 d::Direction = North;
```

Pattern matching can be used with such a data type:

```
1 moveRight(dir::Direction)::Direction =
2   case dir {
3     North → East;
4     East → South;
5     South → West;
6     West → North;
7 }
```

Given a list of such directions, they can be mapped:

```
1 dirs::[Direction] = [ North, South, East, West ];
2 [ moveRight(d) | d ← dirs ]
```

The element types of a union can have components. For example a tree of integers:

```
1 data TreeOfInt =
2   Branch(TreeOfInt,TreeOfInt)
3 | Leaf(Int);
```

Note that a branch term contains two integer trees and a leaf contains an integer. The integers in the tree can be added together:

```
1 addTreeOfInt(t::TreeOfInt)::Int =
2   case t {
3     Branch(l,r) → addTreeOfInt(l) + addTreeOfInt(r);
4     Leaf(n)      → n;
5   }
6 print[Int](addTreeOfInt(Branch(Leaf(100),Leaf(200))));
```

Prints 300. When pattern matching, the case arms are tried in turn, so we can define a short-cut version:

```
1 addTreeOfInt(t::TreeOfInt)::Int =
2   case t {
3     Branch(Leaf(0),t) → addTreeOfInt(t);
4     Branch(t,Leaf(0)) → addTreeOfInt(t);
5     Branch(l,r)       → addTreeOfInt(l) + addTreeOfInt(r);
6     Leaf(n)           → n;
7 }
```

Case arms may contain conditions. For example, adding up all those elements of a tree that satisfy a predicate:

```
1 addIf(p::(Int)→Bool,t::TreeOfInt)::Int =
2   case t {
3     Branch(l,r)      → addIf(p,l) + addIf(p,r);
4     Leaf(n) when p(n) → n;
5     Leaf(n)           → 0;
6   }
7 addIf(fun(n:Int)::Bool n > 100,Branch(Leaf(100),Leaf(200)));
```

Prints 200.

A case-expression can contain any number of values and patterns providing that the number of values and types of the patterns match up. For example:

```
1 sameTree(t1::TreeOfInt,t2::TreeOfInt)::Bool =
2   case t1,t2 {
3     Branch(l1,r1),Branch(l2,r2) →
4       sameTree(l1,l2) and sameTree(r1,r2);
5     Leaf(n1),Leaf(n2) when n1 = n2 → true;
6     t1,t2 → false;
7 }
```

6. Sets and Collection Matching

ESL supports homogenous sets: the empty set `set{}`; a set constructed from an element `e` and a set `s`, `set{e | s}`; a set union `s1 + s2`. Both sets and lists can be processed using pattern matching. For example:

```
1 setContainsInt(s::Set[Int],x:Int)::Bool =
2   case s {
3     set{y | z} when x = y → true;
4     s                   → false;
5 }
```

The operation `setContainsInt` uses pattern matching to select the element `y` which is equal to the argument `x`. Note that sets are unordered, so the pattern `set{y | z}` is *non-deterministic*. When the supplied value `s` matches the pattern, one of the elements in the set is matched against `y`; if the pattern fails to match before the corresponding \rightarrow then a different element `y` will be chosen. In the example, the condition `x = y` is within the scope of the choice, this causes elements from the set `s` to be tried in turn until one that matches `x` is found or the set is exhausted. If all the values have been tried from the set `s` then the case-arm fails and the next case-arm is tried.

Non-deterministic choice may occur more than once in the same pattern. For example, the following returns a duplicate entry in two sets:

```
1 duplicate(s1::Set[Int],s2::Set[Int],notFound:Int)::Int =
2   case s1,s2 {
3     set{x | p}, set{y | q} when x = y → x;
4     s1,s2                           → notFound;
5 }
```

The first case-arm in `duplicate` selects an element `x` from `s1`, then selects `y` from `s2` such that they are equal elements. If there is no element in common then the `notFound` element is returned.

Non-deterministic pattern-based choice can be used over lists using the pattern `p1 + [p2] + p3` where `p1` matches a prefix of a list, `p2` matches some element, and `p3` matches a suffix. The pattern fails in the case that the list is empty. For example, testing whether a given element occurs in a list:

```
1 intMember(n:Int,l:[Int])::Bool =
2   case l {
3     11 + [x] + 12 when x = n → true;
4     l                      → false;
5 }
```

7. Actors

Actors are implemented in terms of *behaviour types* and *behaviours*. A behaviour type provides an interface definition for a set of behaviours. An interface consists of message types and exported identifiers and their types. Consider a simple behaviour type called `Sink` that accepts a single type of message `M(Int)`:

```
1 Act Sink {
2   M(Int);
3 }
```

Any behaviour of type `Sink` must implement the message `M`:

```
1 act sink::Sink {
2   M(n:Int) → print[Str]('got: ' + n);
3 }
```

There can be any number of behaviours with the same type:

```
1 act sink2::Sink {
2   M(n:Int) → {} // Ignore the message.
3 }
```

An actor is created with a specific behaviour:

```
1 s::Sink = new sink;
```

Once created, the actor can receive messages. A message is asynchronous: $s \leftarrow M(100)$ and is placed on the input queue of the target actor. Each actor is processing at most one message at any time, and is idle if there are no messages on its queue and it is not currently handling a message. Since all actors run concurrently, there is no guarantee that of message ordering other than messages sent from the same actor to a target will retain the ordering in which they are sent.

Any number of behaviours can be defined to conform to a behaviour type. The following shows three different implementations of the type `A`. The first, `a`, just loops indefinitely which an actor can do because all message handling occurs concurrently with other actors. The second, `forward` is supplied with a second actor of type `A` to which it forwards messages. Finally, `broadcast` is supplied with a collection of actors of type `A` and will send any message to all of them:

```
1 Act A {
2   M(Int,Bool);
3 }
4
5 act a::A {
6   M(n:Int,b:Bool) →
7     self ← M(n,not(b));
8 }
9
10 actor::A = new a;
11
12 act forward(other::A)::A {
13   M(n:Int,b:Bool) →
14     other ← M(n,b);
15 }
16
17 proxy::A = new forward(actor);
18
19 act broadcast(as:[A])::A {
20   toAll(l:[A],n:Int,b:Bool)::Void =
21     case l {
22       [] → {};
23       h::A:t::[A] → {
24         h ← M(n,b);
25         toAll(t,n,b);
26       }
27     }
28   M(n:Int,b:Bool) →
29     toAll(as,n,b);
30 }
31
32 spray::A = new broadcast([new forward(new a) | i ← 0..100]);
```

Often we want to do some initialisation when an actor is

created. The special message handler `→` is run each time a new actor is created before any messages are processed:

```
1 act sink3::Sink {
2   → print[Str]('I am initialising');
3   M(n:Int) → print[Str]('got: ' + n);
4 }
```

Actors can refer to other actors. The following pair of actors will bounce messages back and forth until the counter runs down:

```
1 Act Ping {
2   Ping(Int,Pong);
3 }
4
5 Act Pong {
6   Pong(Int,Ping);
7 }
8
9 act ping::Ping {
10  Ping(0,p:Pong) → print[Str]('stop');
11  Ping(n:Int,p:Pong) →
12    p ← Pong(n-1,self);
13 }
14
15 act pong::Pong {
16  Pong(0,p:Ping) → print[Str]('stop');
17  Pong(n:Int,p:Ping) →
18    p ← Ping(n-1,self);
19 }
20
21 p1::Ping = new ping;
22 p2::Pong = new pong;
23 p1 ← ping(100,p2);
```

Behaviour definitions can be parameterised so that new actors can initialise the behaviour. For example, in order to provide an identifier to each actor. In the following example, we create a collection of machines and then broadcast jobs. A machine can process a job if the identifier matches:

```
1 Act Machine {
2   Process(Int);
3 }
4
5 act machine(id:Int)::Machine {
6   Process(m:Int) when m = id → {
7     print[Str]('handling job');
8   }
9   Process(m:Int) → {} // Ignore job if not target machine.
10 }
11
12 noOfMs:Int      = 100;
13 noOfJs:Int      = 10000;
14 machines:[Machine] = [ new machine(n) | n ← 0..noOfMs ];
15
16 for job:Int in 0..noOfJs do {
17   let targetMachine:Int = random(noOfMs);
18   in {
19     for m:Machine in machines do {
20       m ← Perform(targetMachine);
21     }
22   }
23 }
```

An actor may perform `become` to change behaviour. The replacement behaviour must implement the same behaviour type. Subsequent messages are handled by the new behaviour. For

example, a resource manager allocates resource on request and becomes pending resource manager for the period of time when resources are unavailable:

```

1 type Resource = ...;
2 Act Consumer {
3   Receive(Resource);
4 }
5
6 Act ResourceManager {
7   Allocate(Consumer);
8   Free(Resource);
9 }
10
11 act manager(rs:[Resource]):ResourceManager {
12   Allocate(c:Consumer) when rs = [] → {
13     become pending;
14     self ← Allocate(c);
15   }
16   Allocate(c:Consumer) → {
17     become manager(tail[Resource](rs));
18     c ← Receive(head[Resource](rs));
19   }
20   Free(r:Resource) → become manager(r:rs);
21 }
22
23 act pending:ResourceManager {
24   Allocate(c:Customer) → self ← Allocate(c);
25   Free(r:Resource) → become manager([r]);
26 }
```

8. Binding and Scope

ESL is a statically scoped language meaning that identifiers can be referenced within the textual binding block in which they are defined. Identifiers can be introduced as: function arguments; let-bindings; letrec-bindings; behaviour definitions; behaviour arguments (equivalent to function arguments); top-level definitions; pattern variables; for-loop controls. This section addresses each of these categories in turn.

Let-expressions introduce local bindings for the scope of the body of the let. A let-expression may have multiple bindings in which case they are performed in parallel:

```

1 let dx:Int = x1-x2;
2   dy:Int = y1-y2;
3 in isqrt((dx*dx)+(dy*dy));
```

A let-binding can be used to define a local function. Special syntax allows a function to be defined without using the keyword `fun`:

```

1 let
2   distance(x1:Int,y1:Int,x2:Int,y2:Int):Float =
3     let dx:Int = x1-x2;
4     dy:Int = y1-y2;
5     in isqrt((dx*dx)+(dy*dy));
6 in distance(100,200,50,70);
```

A letrec-expression can be used to create local recursive functions:

```

1 letrec
2   contains(l:[Int],n:Int):Bool =
3     case l {
4       [] → false;
```

```

5       h:t when h=n → true;
6       h:t → contains(t,n);
7     }
8 in contains([1,2,3,4,5,6,7,8,9],5);
```

A letrec-expression can be used to create mutually recursive function definitions:

```

1 letrec
2   isEven(n:Int):Bool = if n = 0 then true else isOdd(n-1);
3   isOdd(n:Int):Bool = if n = 0 then false else isEven(n-1);
4 in isEven(101);
```

Letrec-bindings that do not establish functions are performed in sequence.

Behaviour definitions contain a collection of bindings that are established as a letrec and scoped over the message handlers of the behaviour. In addition, the names of local bindings can be exported and referenced using the `..` operation. An example, support we have a post office actor that delivers letters to people by matching up the address in the message:

```

1 Act Person {
2   export getAddress():Str;
3 }
4
5 act person(address:Str):Person {
6   export getAddress;
7   getAddress():Str = address;
8   Open(Letter);
9 }
10
11 data Letter = Letter(Str,Str);
12
13 Act PostOffice {
14   Deliver(Str);
15 }
16
17 act postOffice(people:[Person]):PostOffice {
18   findPerson(address:Str,people:[Person]):Person =
19     case people {
20       [] → throw[Person]('cannot deliver to ' + address);
21       p:ps when p.getAddress() = address → p;
22       p:ps → findPerson(address,people);
23     }
24   Deliver(Letter(name,address)) →
25     findPerson(address) ← Open(Letter(name,address));
26 }
```

9. Modules

ESL is a compiled language. The units of compilation are called modules and are usually contained in text files. A module contains a collection of mutually recursive bindings for types, behaviours, functions and values. A module `m` may export some of its bindings so that they can be used by any other module that imports `m`. For example a module defined in a file `a/b/c.esl` defines two names and exports one of them:

```

1 export f;
2
3 f(n:Int):Int = g(n) + 1;
4 g(n:Int):Int = n + 100;
```

A second module can then import `c` and use the exported name:

```

1 export main;
2
3 import 'a/b/c.esl';
4
5 g(n:Int):Int = n = 200;
6
7 act Main:Act{} {
8   → print[Int](f(10));
9 }

```

prints 111. Note that the second module above defines a behaviour names `main`. When building an application there should be a root module that defines `main` which is the entry point for the application. When ESL starts to run a module it creates a single actor with the behaviour `main`. There is no restriction on the behaviour type of `main`.

10. Time

ESL applications can be driven by a system clock. This can be useful when the application is driven by click ticks or when the application should run for a specific length of time. In order to receive clock ticks, a behaviour must implement at least one message handler of type `Time(Int)`. Any actor with such a behaviour will receive clock ticks *when the actor is idle*. If an actor does not define a handler for `ntimeI` then there is no overhead for handling time.

The value of `n` in `Time(n)` is the time in milliseconds since the start of the application. The frequency of the ticks is undefined and therefore there is no guarantee that an actor will receive a tick with a specific value of `n`, and it may be the case the an actor receives multiple messages with the same value of `n`.

An actor may call `wait(n)` which causes the actor's thread to wait for `n` milliseconds. Since all actors have concurrent behaviours, a call of `wait` will not affect any other actor.

The following definition shows a typical pattern involving click ticks: two message handlers are defined, the first to detect whether a limit has been achieved and stops the application, the second just ignores the click:

```

1 Time(n:Int) when n > limit → stopAll();
2 Time(n:Int) → {}

```

11. State

A value binding establishes an association between an identifier and a value. The association has a scope that defines the ESL code where the identifier can be referenced. ESL uses *lexical scoping* where the association is established in a construct (module, behaviour, let, letrec, function, case-arm, message handler) whose textual definition is the scope.

The original definition of the actor model of computation does not support side-effects. This restriction provides a rather austere application development platform where certain obvious implementation approaches become more unwieldy than they might otherwise be. Therefore, ESL provides side-effects

on value bindings. These must be used with care since the actor model was originally designed to avoid issues such as race conditions that occur because of shared state.

Side-effects may be used to provide actors with mutable state:

```

1 Act BankAccount {
2   Deposit(Int);
3 }
4
5 act account::BankAccount {
6   funds:Int = 0;
7   Deposit(n:Int) → funds := funds + n;
8 }

```

An actor may encapsulate its state. Since an actor is singly-threaded, this means that there can be no doubt regarding whether an update has occurred or not when the value of a state variable is used, *i.e.* the state is not used by multiple threads. However, it is often useful for an actor to provide access to its internal state via its interface. This can be achieved in one of two key ways: message passing or an interface operation.

If access is provided by message passing then reference to the state remains singly threaded, however the message must contain the requesting actor who receives the state value via a return message:

```

1 Act BankAccount {
2   Deposit(Int);
3   Withdraw(Int,A);
4 }
5
6 act account::BankAccount {
7   funds:Int = 0;
8   Deposit(n:Int) → funds := funds + n;
9   Withdraw(n:Int,a::A) when n >= funds → {
10     funds := funds - n;
11     A ← Withdrawn;
12   }
13   Withdraw(n:Int,a::A) →
14     a ← WithdrawFailed;
15 }

```

Alternatively, access may be provided via an interface operation. In this case the state can be accessed using multiple threads raising the possibility of race conditions. Consider the following version of the bank account:

```

1 Act BankAccount {
2   export
3     getFunds():Int → Int;
4     withdraw:Int → Void;
5     deposit:Int → Void;
6 }
7
8 act account::BankAccount {
9   export getFunds,withdraw,deposit;
10  funds:Int = 0;
11  getFunds():Int = funds;
12  deposit(n:Int):Void = funds := funds + n;
13  withdraw(n:Int):Void =
14    if n >= funds
15      then funds := funds - n;
16    else {}
17 }

```

Two different actors may share access to a bank account. The

first actor performs the following:

```
1 a.deposit(100);
2 a.withdraw(10);
3 print[Str]('funds = ' + a.getFunds());
```

what value will be printed? We might expect the funds to be increased by 90. However, the answer depends on whether the second actor has performed a deposit or withdrawal in between line 1 and 2 or between line 2 and 3.

In order to be sure, such transactional blocks must be protected. All access to shared state must use locks to gain exclusive access. Each actor that uses the exported operations must do so within a `grab(lock)` block. Any value may be used as the lock providing the same value is used by all threads that share the state. In this case it makes sense to use the account as the lock:

```
1 grab(a) {
2   a.deposit(100);
3   a.withdraw(10);
4   print[Str]('funds = ' + a.getFunds());
5 }
```

Now providing that all actors that share the account use a similar `grab(a)` to wrap any transactions, the actor above can be sure that the funds will have been increased by 90.

The scheme above relies on the client actors being well behaved in their use of the bank account by wrapping transactions in a `grab(a)`. A better way is for the bank account to enforce the protection of the state whilst still offering an interface. This can be achieved by passing the account a transaction function as follows:

```
1 Act BankAccount {
2   export
3     transaction::()→Int, (Int)→Void, (Int)→Void→Void
4 }
5
6 act account::BankAccount {
7   export transaction;
8   funds:Int = 0;
9   getFunds():Int = funds;
10  deposit(n:Int):Void = funds := funds + n;
11  withdraw(n:Int):Void =
12    if n >= funds
13      then funds := funds - n;
14    else {}
15  transaction(action::() → Int,
16             (Int) → Void,
17             (Int) → Void) → Void:Void =
18    grab(self) {
19      action(getFunds,deposit,withdraw);
20    }
21 }
```

Any client actor must provide the account with an action which is performed on the client's thread and is guaranteed to occur within the scope of the lock:

```
1 a.transaction(fun(getFunds::() → Int,
2                   deposit::(Int) → Void,
3                   withdraw::(Int) → Int):Void {
4   deposit(100);
5   withdraw(10);
6   print[Str]('funds = ' + getFunds());
```

7});

12. Concurrency

ESL actors run concurrently. An actor is created using `new b` or `new b(x,...)` which immediately returns a handle to the newly created actor and schedules the thread of the newly created actor. The behaviour `b` may have an initialisation clause (a message handler starting with `→`) that is the first action performed on the new thread. After initialisation, the new thread enters a loop that inspects the actor's message queue and dispatches to a message handler in `b` if a message exists, otherwise the thread waits until a message is received. If `b` defines a handler for `Time` (`Int`) then the ESL system sends the actor a message providing the queue is empty.

ESL is designed to make the creation and interaction of concurrent actors very lightweight. As described in `sec:state`, actors can share state and, if so, must make appropriate use of locks. However, in general the state of an actor is encapsulated and is modified by third party actors using message passing.

The use of large scale concurrency and asynchronous message passing changes the conventional approach to system design and implementation which is based on task decomposition and sequential ordering. This section provides a number of examples that show how actors can be used to implement tasks that take advantage of the concurrent computational model.

12.1. Search

Actors make the implementation of applications involving brute-force search simple because a number of actors can be created that concurrently navigate the search space. The amount of search space explored at any given time is proportional to the number of actors created (and the amount of processing power available).

Consider a square two-dimensional string array that contains a single occurrence of `'*'` at an unknown location. A sequential program might start at `(0,0)` and move through the locations until it finds `'*'`. The worst case execution will inspect all the locations in the array.

A simple actor-based solution creates several actors that all search for the `'*'` concurrently. Each actor can start at a random location and move around at random. This provides a way of comparing the benefits of scaling up the number of actors in a brute-force search.

Figure 3 shows the implementation of the simple search application. The worst-case results of several runs are as follows:

numOfSearchers	count
1	2155079
10	40255
100	30331
1000	656

```

1 Act Main      { }
2 Act Searcher   { Time(Int);      }
3 Act Controller { Time(Int); Found; }
4
5 size:Int          = 1000;
6 numofSearchers:Int = 10;
7 board:Array[Array[Str]] = new Array[Array[Str]](size);
8 max(a:Int,b:Int):Int = if a > b then a; else b;
9
10 act controller:Controller {
11   count:Int = 0;
12   Time(n:Int) → count := count + 1;
13   Found → {
14     print[Str]('Found in ' + count + ' steps');
15     stopAll();
16   }
17 }
18
19 act searcher(control:Controller):Searcher {
20   x:Int = random(size);
21   y:Int = random(size);
22   delta(n:Int):Int = (n + max(random(3) - 1,0)) % size;
23   Time(n:Int) →
24     if board[x][y] = '*'
25     then control ← Found;
26     else { x := delta(x); y := delta(y); }
27 }
28
29 act main:Main {
30   → {
31     for x:Int in 0..size do {
32       board[x] := new Array[Str](size);
33       for y:Int in 0..size do
34         board[x][y] := '';
35     }
36   board[random(size)][random(size)] := '*';
37   let control:Controller = new controller;
38   in {
39     for i:Int in 0..numofSearchers do
40       new searcher(control);
41   }
42 }
43 }
```

Figure 3: Searching an Array

12.2. Quicksort

Sequential quicksort was defined in section 4. The algorithm clearly lends itself to concurrency since it splits on an element such that the lists of elements below and above the chosen number can be sorted independently. Figure 4 shows the definition of concurrent quicksort in ESL. The actor `qsorтер` selects an element `x` and creates two independent child `qsorтер` actors to sort the list of elements below `x` and above `x` respectively before changing behaviour to a `qwaiter`.

The `qwaiter` actor implements a typical scenario: waiting for two independent calculations to terminate before continuing. It achieves this using `Left` and `Right` tokens in the two children and waiting until it receives both tokens before sending the sorted list to its own parent.

Figure 5 shows a diagram representation of a simple quicksort in terms of the actors, their links and the messages that are sent. Note the dashed arrow showing `qsorтер` actors becoming `qwaiter` actors.

```

1 data Direction = Left | Right | Final;
2
3 Act QSort { Sorted([Int],Direction); }
4
5 nums:[Int] = [ random(50) | n:Int ← 0..25 ];
6
7 act qmain(l:[Int]):QSort {
8   → new qsorтер(self,l,Final);
9   Sorted(l:[Int],Final) → {
10     print[Int](l);
11     stopAll();
12   }
13 }
14
15 act qwaiter(parent:QSort,n:Int,dir:Direction):QSort {
16   left:[Int] = null[[Int]];
17   right:[Int] = null[[Int]];
18   check():Void =
19     if left <> null[[Int]] and right <> null[[Int]]
20     then parent ← Sorted(left+n+right,dir);
21     else {}
22   Sorted(l:[Int],Left) → { left := l; check(); }
23   Sorted(l:[Int],Right) → { right := l; check(); }
24 }
25
26 act qsorтер(parent:QSort,l:[Int],dir:Direction):QSort {
27   → case l {
28     [] → parent ← Sorted(l,dir);
29     x:Int:l:[Int] → {
30       new qsorтер(self,[ n | n:Int ← 1, ?(n < x), Left]);
31       new qsorтер(self,[ n | n:Int ← 1, ?(n > x), Right]);
32       become qwaiter(parent,x,dir);
33     }
34   }
35 }
36 Sorted(l:[Int],d:Direction) → throw[Void] 'error!';
37 }
38
39 act main:Act{} { → new qmain(nums); }
```

Figure 4: Concurrent Quicksort

12.3. Termites

Actor based systems can be used to exhibit *emergent behaviour* where system-level behaviour can be observed as a result of many simple individual behaviours even though each individual behaviour has no knowledge of the whole. An example is *termites* shown in figure 6 where a collection of twigs (black squares) and moved by a collection of termites (red squares) to form piles even though the individual termites do not have any knowledge of pile-formation.

The behaviour of a termite is simple: walk around at random until a twig is found that is not next to other twigs. The termite then picks up the lone-twig and walks at random until it finds another twig that is next to other twigs. The termite then drops the twig and then starts over again.

The implementation of termites in ESL is shown in figure 7. The implementation shows a typical scenario in actor-based systems where there is a single world-state and many different actors that share the state. Each individual needs the state in order to determine its next move which involves an update to the world state. Given that all the actors are operating concurrently,




Figure 5: Example Quicksort





Figure 6: Termites Before and After

this dependency on a single world state can cause problems if access and updates are not protected. Conventional programming languages provide *data locks* to deal with shared state.

ESL provides locks, but often they are not required because actors operate in terms of asynchronous messages and queues. The diagram below shows how this works:



The world at the centre of the figure represents the shared state. Four independent worker actors are shown, although any number can be involved. The shared state offers an interface that supports the operations required by the worker actors which, in turn, provide an interface that represents the different outcomes from performing the world operations.


From the perspective of a single worker, it looks like they

get dedicated access to the shared world state because a message sent to the world is queued until the world becomes free. Each worker message must include the worker actor so that the shared state can reply to it. When a worker sends a message to the shared state the worker becomes idle and will be woken up by the reply from the shared state. When the shared state processes a worker message, the worker has exclusive access to the data, the data can be changed and then the message is sent back to the worker which is woken up.


12.4. Dining Philosophers

Dining Philosophers is a standard scenario that is used to exemplify the issues of shared access to resources. Several concurrent processes (philosophers) are vying for shared resources (chopsticks) in order to eat. Each process needs exactly two resources to each and each resource is shared by exactly two processes. When a process acquires a pair of resources, it performs activity (eats) and then releases the resources. Access to resources is assumed to be fair. If a process acquires resources sequentially, then a situation can arise where all resources have acquired one resource and is awaiting on a resource that has been acquired by another process (starvation). In order to avoid starvation, the processes need to wait until they can individually acquire a pair of resources as an atomic action.

A situation where the philosophers are happily eating is shown below:



In the case where chopsticks are grabbed one at a time, the situation rapidly deadlocks:



The implementation of deadlock-free dining philosophers is shown below:

```

1 Act Main { Time(Int); }
2 Act Grid { SetColour(Int,Int,Str); TermiteAt(Int,Int,Int); }
3
4 size:Int = 120;
5 limit:Int = 40000;
6 numOfTermites:Int = 600;
7 twig:Str = 'black';
8 background:Str = 'white';
9 grid:Grid = new 'esl.grid.Grid'[Grid](size,size,6);
10
11 isLegal(x:Int,y:Int):Bool =
12   (x >= 0 and x < size) and (y >= 0 and y < size);
13
14 Act Termite { Search; Drop; FindSpace; GetAway(Int); }
15
16 act termite(id:Int,w:World):Termite {
17
18   x:Int = random(size);
19   y:Int = random(size);
20   dx:Int = random(3) - 1;
21   dy:Int = random(3) - 1;
22
23   randomDir():Void = {
24     dx := random(3)-1;
25     dy := random(3)-1;
26   }
27   move():Void = {
28     x := (x + dx); y := (y + dy);
29     if x < 0
30       then { dx := 1; x := 0; move(); }
31     else if x > (size - 1)
32       then { dx := -1; x := (size - 1); move(); }
33     else if y < 0
34       then { dy := 1; y := 0; move(); }
35     else if y > (size - 1)
36       then { dy := 0 - 1; y := (size - 1); move(); }
37     else grid ← TermiteAt(id,x,y);
38   }
39   moveRandom():Void = {
40     randomDir();
41     move();
42   }
43
44 → self ← Search;
45
46 Search → {
47   moveRandom();
48   w ← TryPickup(x,y,self);
49 }
50 Drop → {
51   moveRandom();
52   w ← FindPile(x,y,self);
53 }
54 FindSpace → {
55   moveRandom();
56   w ← TryDrop(x,y,self);
57 }
58 GetAway(0) → {
59   self ← Search;
60 }
61 GetAway(n:Int) → {
62   move();
63   self ← GetAway(n-1);
64 }
65 }
66
67
68 Act World {
69   TryPickup(Int,Int,Termite);
70   FindPile(Int,Int,Termite);
71   TryDrop(Int,Int,Termite);
72 }
73
74 act world:World {
75   locations:Array[Array[Str]] =
76     let a:Array[Array[Str]] = new Array[Array[Str]](size);
77     in {
78       for x:Int in 0..size do {
79         a[x] := new Array[Str](size);
80         for y:Int in 0..size do {
81           a[x][y] := if random(100) < 30
82             then twig
83             else background;
84           grid ← SetColour(x,y,a[x][y]);
85         }
86       }
87       edb.display[Grid]('Termites',grid);
88       a;
89     }
90   termites:[Termite] =
91     [ new termite(n,self) | n:Int ← 0..numOfTermites ];
92   foundSingleton(x:Int,y:Int):Bool =
93     locations[x][y] = twig and twigCount(x,y) < 5;
94   foundPile(x:Int,y:Int):Bool =
95     locations[x][y] = twig and twigCount(x,y) > 4;
96   isTwig(x:Int,y:Int):Bool =
97     if isLegal(x,y)
98       then locations[x][y] = twig;
99       else false;
100  twigCount(x:Int,y:Int):Int =
101    sum([ if isTwig(x+dx,y+dy) then 1 else 0 |
102        dx ← [-1,0,1],
103        dy ← [-1,0,1],
104        ?(x<>0 or y<>0) ]);
105
106 TryPickup(x:Int,y:Int,t:Termite) → {
107   if foundSingleton(x,y)
108     then {
109       locations[x][y] := background;
110       grid ← SetColour(x,y,background);
111       t ← Drop;
112     } else t ← Search;
113   }
114 FindPile(x:Int,y:Int,t:Termite) → {
115   if foundPile(x,y)
116     then t ← FindSpace;
117     else t ← Drop;
118   }
119 TryDrop(x:Int,y:Int,t:Termite) → {
120   if locations[x][y] = background
121     then {
122       locations[x][y] := twig;
123       grid ← SetColour(x,y,twig);
124       t ← GetAway(20);
125     } else t ← FindSpace;
126   }
127 }
128
129 w:World = new world;
130
131 act main:Main {
132   Time(n:Int) when n > limit → stopAll();
133   Time(n:Int) → { }
134 }
```

Figure 7: Termites

```

1 number_of_philosophers:Int = 10;
2 eat_time:Int = 50;
3 think_time:Int = 50;
4
5 act Main { Time(Int); }
6 act Chopstick { }
7 act Philosopher { Time(Int); }
8
9 eat():Void = wait(eat_time);
10 think():Void = wait(think_time);
11
12 act philosopher(i:Int,
13   left:Chopstick,
14   right:Chopstick):Philosopher {
15   Time(n:Int) → {
16     think();
17     grab(left,right) {
18       eat();
19     }
20   }
21 }
22
23 act chopstick:Chopstick {}
24
25 chopsticks:[Chopstick] =
26   [ new chopstick | i:Int ← 0..number_of_philosophers ];
27 chop(i:Int):Chopstick =
28   nth[Chopstick](chopsticks,i%number_of_philosophers);
29
30 philosophers:[Philosopher] =
31   [ new philosopher(i,chop(i),chop(i+1))
32   | i:Int ← 0..number_of_philosophers ];

```

Deadlock is avoided because the `grab` block acquires the locks on `left` and `right` at the same time. If the locks are not available then the `grab` block will wait until they are both free and acquire them simultaneously. Changing the definition of `philosopher` to the following leads to deadlock:


```

1 act philosopher(i:Int,
2   left:Chopstick,
3   right:Chopstick):Philosopher {
4   Time(n:Int) → {
5     think();
6     grab(left) {
7       grab(right) {
8         eat();
9       }
10    }
11  }
12 }


```

12.5. Segregation

Thomas Schelling proposed an agent-based model of segregation in *Thomas C Schelling. Models of Segregation. American Economic Review, 59(2):488–493, 1969.* which demonstrates that relatively mild preferences for neighbours of a similar race can lead to high levels of segregation. The Schelling model shows how a completely random mix of two types of independent individuals:



can lead to a structured outcome:



when none of the individuals would claim to have designed the outcome. This is another example of emergent behaviour, where relative simple individual behaviours can lead to a combined outcome that none of the individuals have planned for.

The data types used in the ESL implementation of segregation are shown below:

```

1 type Main = Act { }
2
3 type Agent = Act {
4   export getX():→Int;
5   getY():→Int;
6   setX:(Int)→Void;
7   setY:(Int)→Void;
8 }
9
10 type Grid = Act {
11   SetColour(Int,Int,Str);
12   Swap(Int,Int,Int);
13 }
14
15 type Population = Act {
16   Move(Agent);
17 }
18
19 data Loc = Coord(Int,Int);
20 coordX(p:Loc):Int = case p { Coord(x:Int,y:Int) → x; }
21 coordY(p:Loc):Int = case p { Coord(x:Int,y:Int) → y; }

```

As we see below, an agent is just a passive structure with state. Each agent is singly threaded through the shared population state by a series of messages. A grid is a foreign actor implemented in Java which draws a colour at an (x,y) position and can swap the colours between two different locations. A population controls the grid.

An agent manages its state and initiates a sequence of messages to a population:

```

1 act agent(x:Int,y:Int)::Agent {
2
3   export getX,getY,setX,setY;
4
5   getX():Int = x;
6   getY():Int = y;
7   setX(x0:Int)::Void = x := x0;
8   setY(y0:Int)::Void = y := y0;
9
10  → population ← Move(self);
11
12 }
```

The following parameters and constants control the segregation application:

```

1 diffLimit:Float = 25.00; // % comfort limit.
2 width:Int = 1000; // width of population.
3 height:Int = 600; // height of population.
4 redpc:Int = 40; // % of population red.
5 emptypc:Int = 10; // % of population empty.
6 empty:Int = 0; // constant representing empty.
7 red:Int = 1; // constant representing red.
8 blue:Int = 2; // constant representing blue.
9 vLength:Int = 0; // number of vacancies.
10
11 opp(c:Int)::Int = if c = red then blue; else red;
12 colour(c:Int)::Str =
13   if c = red then 'red';
14   else if c = empty then 'white';
15   else 'blue';
16
17 legalx(x:Int)::Bool = (x >= 0) and (x < width);
18 legaly(y:Int)::Bool = (y >= 0) and (y < height);
```

The main work of the application is performed by the population actor:

```

1 act pop::Population {
2
3   createAgent(x:Int,y:Int)::Agent = {
4     grid ← SetColour(x,y,colour(population[x][y]));
5     new agent(x,y);
6   }
7
8   createVacancy(x:Int,y:Int)::Loc = {
9     grid ← SetColour(x,y,colour(empty));
10    Coord(x,y);
11  }
12
13  population:Array[Array[Int]] =
14    // Create the basic population...
15  let a:Array[Array[Int]] = new Array[Array[Int]](width);
16  in {
17    for w:Int in 0..width do {
18      a[w] := new Array[Int](height);
19      for h:Int in 0..height do {
20        a[w][h] :=
21          probably(100-emptypc)::Int {
22            probably(redpc)::Int
23              red;
24              else blue;
25            } else empty;
26        }
27      a;
28    }
29 }
```

```

31 agents::Array[Array[Agent]] =
32   // Create the agents...
33  let a::Array[Array[Agent]] =
34    new Array[Array[Agent]](width);
35  in {
36    for x:Int in 0..width do {
37      a[x] := new Array[Agent](height);
38      for y:Int in 0..height do {
39        a[x][y] := if population[x][y] <> empty
40          then createAgent(x,y)
41          else null[Agent];
42      }
43    }
44  a;
45 }
46
47 createVacancies()::Array[Loc] =
48   // create an array of vacancies for easy access...
49  let vacancies::[Loc] = [ createVacancy(x,y) |
50    x:Int ← 0..width,
51    y:Int ← 0..height,
52    ?(population[x][y]=empty) ];
53  in {
54    vLength := length[Loc](vacancies);
55    let v:Array[Loc] = new Array[Loc](vLength);
56    in {
57      for i:Int in 0..vLength do {
58        v[i] := nth[Loc](vacancies,i);
59      }
60    v;
61  }
62
63 vacancies::Array[Loc] = createVacancies();
64
65 popSet(x:Int,y:Int,c:Int)::Void = population[x][y] := c;
66
67 diffCellCount(x:Int,y:Int,c:Int)::Int = length[Int]([ 1 |
68   dx ← [-1,0,1],
69   dy ← [-1,0,1],
70   ?not(dx=0 and dy=0),
71   ?legalx(x+dx),
72   ?legaly(y+dy),
73   ?population[x+dx][y+dy] = c]);
74
75 diffpc(x:Int,y:Int)::Float =
76   (intToFloat(diffCellCount(x,y,opp(population[x][y]))) /
77   8.0)*100.0;
78
79 Move(a:Agent) → {
80  let x:Int = a.getX(); y:Int = a.getY(); in
81  if diffpc(x,y) > diffLimit
82  then
83    let i:Int = random(vLength); in
84    let p:Loc = vacancies[i]; in
85    let x0:Int = coordX(p); y0:Int = coordY(p);
86    in {
87      vacancies[i] := Coord(x,y);
88      popSet(x0,y0,population[x][y]);
89      popSet(x,y,empty);
90      grid ← Swap(x,y,x0,y0);
91      a.setX(x0);
92      a.setY(y0);
93      self ← Move(a);
94    }
95    else self ← Move(a);
96  }
97 }
```



A pop actor receives a message `Move(a)`. Since the receiver is singly threaded, this provides `a` with exclusive access to the

arrays `population` and `vacancies`. If each agent shared access to these resources, it is likely that movements could interleave and the state could become inconsistent. Once the move has occurred, the `pop` actor sends itself a `Move(a)` message which is queued with all other move requests. Since messages are delivered fairly, no agent is starved of movement.


12.6. Services

In a free-market economy commercial services can be traded freely without restrictions. This might lead to a conclusion that, starting in a situation where wealth is evenly distributed, freely trading services will not change the distribution. However, as shown in the *MSc Thesis by Roohi Bharti: HIVE - An Agent Based Modeling Framework*, free trade over time leads to a situation where wealth moves to a very small number of individuals.

Suppose that we have 100 traders each with £100. The following display shows the traders where their wealth is represented by the size of the dot:



When the simulation is run for a long time, many traders are forced out of business. Random selection means that some traders are successful consumers more often than they are providers which places them at a disadvantage:



Trading involves a service being passed between a requester and a provider. Traders can be both service requesters and providers. Both the requester and the provider must not be involved in another transaction. The requester initiates the transaction, which may be refused by the selected trader. When the request is accepted then the transaction takes place and costs £3 which is immediately transferred from the requester to the provider. After the transaction, both traders are free to perform other transactions. Even though providers are selected at random, after a short while some traders get an advantage:

The implementation of services is shown in figure 8. A display is generated as a square table base don the square root of the number of traders. The size of the circle shown for each trader is scaled based on the total amount of money available in the system. Since the simulation runs for a long time, the display is not updated every cycle to prevent flickering.

13. Polymorphism

When a function is defined, its type signature defines the types of its arguments and its return type. Consider the function `idInt`:

```
1 idInt(x:Int)::Int = x;
```

which is similar to the function `idBool`:

```
1 idBool(x:Bool)::Bool = x
```

which is similar to many other functions such as `idFloat`, `idStr` and `idListOfInt`. In all cases the particular type of the argument

```

1 Act Main { Time(Int); }
2
3 Act Customer {
4   export picture():Display;
5   Inc; Dec;
6 }
7
8 Act Facilitator { PairUp(Trader); }
9
10 act facilitator::Facilitator {
11
12   pending:[Trader] = [];
13   count:Int = 0;
14   root:Int = round(sqrt(noOfTraders));
15   rowTraders(n:Int):[Trader] = take[Trader](drop[Trader](traders,n*root),root);
16   row(n:Int):Row = Row([], [ Data([],c.picture()) | c < rowTraders(n) ]);
17   table():Display = Table([], [ row(n) | n < 0..root ]);
18   chooseTrader():Trader = nth[Trader](pending,random(length[Trader](pending)));
19
20   PairUp(c:Trader) when pending = [] → pending := c:pending;
21   PairUp(c1:Trader) → {
22     let c2:Trader = chooseTrader();
23     in {
24       c1 ← Inc;
25       c2 ← Dec;
26       pending := remove[Trader](c2,pending);
27       count := count + 1;
28       if count % (20 * noOfTraders) = 0
29         then edb ← Show('Services', table());
30       else {}
31     }
32   }
33 }
34
35 theFacilitator::Facilitator = new facilitator;
36
37 act trader(money:Int):Trader {
38   export picture;
39   wealth:Int = money;
40   picture():Display = Picture(60,60,[Circle(30,30,round(((wealth*20.0)/(1.0*totalMoney)) * 30.0),#)]);
41   → theFacilitator ← PairUp(self);
42   Inc → {
43     wealth := wealth + 3;
44     theFacilitator ← PairUp(self);
45   }
46   Dec when wealth > 3 → {
47     wealth := wealth - 3;
48     theFacilitator ← PairUp(self);
49   }
50   Dec → become emptyTrader;
51 }
52
53 act emptyTrader:Trader {
54   export picture;
55   picture():Display = Picture(60,60,[]);
56   Inc → {}
57   Dec → {}
58 }
59
60 limit:Int = 300000;
61 noOfTraders:Int = 100;
62 initialFunds:Int = 100;
63 totalMoney:Int = noOfTraders * initialFunds;
64 traders:[Trader] = [ new trader(initialFunds) | i < 0..noOfTraders ];
65
66 act main::Main {
67   Time(n:Int) when n < limit → {}
68   Time(n:Int) → stopAll();
69 }

```

Figure 8: Services

is not relevant since the function always does the same thing: return the argument. It is important that the return type of the function is the same as the type of the argument. ESL allows a polymorphic function to be defined as follows:

```
1 id[T] (x::T)::T = x;
```

When such a definition is used, the particular types must be supplied:

```
1 n::Int = id[Int](100);
2 b::Bool = id[Bool](true);
```

Multiple type arguments may be used. For example, the following is a function that applies a function f to all elements of a list in order to transform them from type `Source` to type `Target`:

```
1 map[Source,Target] (l::[Source],f::(Source)→Target)::[Target]=
2   case l {
3     [] → [];
4     h:t → f(h) : map[Source,Target](t,f);
5   }
```

Type functions provide abstractions over type definitions. The type function `ListOf` maps a type T to the type $[T]$:

```
1 type ListOf[T] = [T];
```

Once defined, `ListOf[T]`, for some specific type expression T can be used as a type.

Data types can be defined parametrically. For example the following is a tree type for any type T :

```
1 data Tree[T] =
2   Branch(Tree[T],Tree[T])
3 | Leaf(T);
```

Since the constructors `Branch` and `Leaf` are defined in the context of a type function, type arguments must be supplied when the constructors are used in patterns or to construct data values. For example:

```
1 treeTrans[T1,T2] (map::(T1) → T2,t::Tree[T1]):Tree[T2] =
2   case t {
3     Branch[T1](l,r) → Branch[T2](treeTrans[T1,T2](map,l),
4       treeTrans[T1,T2](map,r));
5     Leaf[T1](v)      → Leaf[T2](map(v));
```

and the two data values:

```
1 t1::Tree[Int] =
2   Branch[Int](Leaf[Int](100),Leaf[Int](200));
3 t2::Tree[Bool] =
4   treeTrans[Int,Bool](fun(i:Int)::Bool i > 100,t1);
```

The type mechanisms are hidden behind the language structures. If these are unpacked then the reason for the type arguments should be clear:

```
1 type Tree = rec X. Fun[Y]
2   union {
3     Branch(X[Y],X[Y]),
4     Leaf(Y)
5   }
```


The constructor types in a union-type definition are viewed as function that inject into the union-type. Since the union-type

is defined as a type function, the constructors must be supplied with a type whenever they are used.

13.1. Map/Reduce

Map/reduce is an approach to processing large amounts of data that involves transforming the data and grouping with respect to keys. It is a pattern that is particularly suited to large amounts of concurrent processing because the mapping can occur independently and the grouping can be divided between different processing units. The approach has been used to help process internet data, such as counting the number of occurrences of text.

The diagram below shows how map/reduce works:



The data is of type x and is mapped to produce a collection of key-value pairs where the keys are of type K and the value of type V . Many different pairs may contain the same key; the pairs are grouped so that all the values associated with the same key are located together. Finally, each collection of values for the same key are reduced to produce a value of type y .

The map/reduce process is independent of the type of data x , the type of the keys K and the type of the target values y . It is therefore a suitable candidate for polymorphism. The example will use a collection of words:

```
1 import 'esl/strings.esl';
2
3 words::[Str] = splitBy(#\space,'Sed ut perspiciatis unde
omnis iste natus error sit voluptatem accusantium
doloremque laudantium, totam rem aperiam, eaque ipsa
quae ab illo inventore veritatis et quasi architecto
beatae vitae dicta sunt explicabo. Nemo enim ipsam
voluptatem quia voluptas sit aspernatur aut odit aut
fugit, sed quia consequuntur magni dolores eos qui
ratione voluptatem sequi nesciunt. Neque porro
quisquam est, qui dolorem ipsum quia dolor sit amet,
consectetur, adipisci velit, sed quia non numquam eius
modi tempora incidunt ut labore et dolore magnam
aliquam quaerat voluptatem. Ut enim ad minima veniam,
quis nostrum exercitationem ullam corporis suscipit
laboriosam, nisi ut aliquid ex ea commodi consequatur?
Quis autem vel eum iure reprehenderit qui in ea
voluptate velit esse quam nihil molestiae consequatur,
vel illum qui dolorem eum fugiat quo voluptas nulla
pariatur?');
```

We will define a map/reduce implementation that uses polymorphic ESL actors to perform concurrent maps. The single implementation will be used in two ways: (1) to produce the number of words of each length; (2) to produce the number of occurrences of each word.

```
1 Act Main {}
```

```
2
```

```

3 data KV[Key,Value] = Key(Key,Value);
4
5 type Map[In] = Act { Perform(In); }
6 type Group[Key,Value] = Act { Store(KV[Key,Value]); }
7 type Reduce[Key,Value] = Act { Process(KV[Key,[Value]]); }

```

The data type `kv` is polymorphic with respect to the type of the key and the value. The type `Map` is a behaviour that it independent of the type of data items it processes via the `Perform` message. The type `Group` stores key-value pairs independent of they specific types, and the type `Reduce` processes key-value pairs after they have been grouped.

A map actor is created with the behaviour `mkMap`:

```

1 act mkMap[In,Key,Value](f:(In)→KV[Key,Value],
2                         g:Group[Key,Value]):Map[In] {
3     Perform(x:In) →
4         g ← Store(f(x));
5 }

```

The `mkMap` function takes a mapping function `f` and a grouping actor `g`. When a map actor is sent a `Perform(x)` message it uses `f` to transform `x` to a key-value pair which is supplied to the grouping actor `g`.

A group actor has the following behaviour:

```

1 act mkGroup[K,V](n:Int,reduce:Reduce[K,V]):Group[K,V] {
2     results :: [KV[K,[V]]] = [];
3     add(x:KV[K,V],rs:[KV[K,[V]]]):[KV[K,[V]]] =
4         case x {
5             Key[K,V](i::K,v::V) →
6                 case rs {
7                     [] → [Key[K,V](i,[v])];
8                     Key[K,V](j,vs):rs when i = j →
9                         Key[K,V](i,v:vs):rs;
10                    r:rs → r:add(x,rs);
11                }
12            }
13        Store(x:KV[K,V]) when n = 1 → {
14            results := add(x,results);
15            for r:KV[K,V] in results do {
16                reduce ← Process(r);
17            }
18        }
19        Store(x:KV[K,V]) → {
20            results := add(x,results);
21            n := n - 1;
22        }
23 }

```

A group actor is supplied with the number of independent mappings `n` and a reducer actor `reduce`. The grouped results are maintained in the list `results`. Note that the type of `results` is `[KV[K,[V]]]`; the idea is that each key is associated with all the values supplied to the grouping agent.

A grouping agent is supplied with key-value pairs in the message `Store(KV(k,v))`. If this not the last pair, then it is added to the results using `add` which ensures that all the values associated with the same key are grouped together. When the last key-value pair is received, the results are passed to the reducing actor.

Map/reduce is performed by the following function:

```

1 mapReduce[In,Key,Value](l:[In],
2                         f:(In)→KV[Key,Value],
3                         g:Group[Key,Value]):Void =

```

```

4     for x:In in l do {
5         (new (mkMap[In,Key,Value])(f,g)) ← Perform(x);
6     }

```

13.1.1. All Words of a Given Length

```

1 act reduceWordLen:Reduce[Int,Str] {
2     Process(r:KV[Int,[Str]]) →
3         case r {
4             Key[Int,[Str]](i:Int,s:[Str]) →
5                 print[Str](i + ' ' + s);
6         }
7 }
8
9 strLen(s:Str):KV[Int,Str] =
10    Key[Int,Str](length[Int](s.explode),s);
11
12 mapReduceWordLength():Void =
13    let r:Reduce[Int,Str] = new reduceWordLen;
14    n:Int = length[Str](words); in
15    let g:Group[Int,Str] = new (mkGroup[Int,Str])(n,r);
16    in mapReduce[Str,Int,Str](words,strLen,g);

```

13.1.2. Occurrences of a Word

```

1 act occurs:Reduce[Str,Int] {
2     Process(r:KV[Str,[Int]]) →
3         case r {
4             Key[Str,[Int]](s:Str,is:[Int]) →
5                 print[Str](s + ' occurrences = ' + length[Int](is));
6         }
7 }
8
9 strOccurs(s:Str):KV[Str,Int] = Key[Str,Int](s,1);
10
11 mapReduceOccurrences():Void =
12    let r:Reduce[Str,Int] = new occurs;
13    n:Int = length[Str](words); in
14    let g:Group[Str,Int] = new (mkGroup[Str,Int])(n,r);
15    in mapReduce[Str,Str,Int](words,strOccurs,g);

```

13.2. Cached Functions

Single-argument functions without side effects can be cached so that the computation needed to produce the result is performed once and then added to a table. Subsequent applications of the function to the same argument will just look the value up in the table. Although the type of the argument and the return value differs between functions, the caching activity is the same in all cases. polymorphism can be used to abstract the type of the argument and return value:

```

1 cache[Arg,Value](f:(Arg) → Value):(Arg) → Value =
2     let table:Hash[Arg,Value] = new Hash[Arg,Value];
3     in {
4         fun(a:Arg):Value
5             if table.hasKey(a)
6                 then table.get(a);
7             else
8                 let v:Value = f(a);
9                 in {
10                     table.put(a,v);
11                     v;
12                 }
13     }

```

Suppose we construct lists of the form `0..n` repeatedly for a range of values `n` where the same `n` may occur frequently. It is wasteful to construct the same list multiple times, especially if the value of `n` is high. It is much more efficient to use cache:

```
1 getList::(Int) → [Int] = cache[Int,[Int]](fun(n:Int) 0..n);
```


14. Object Orientation

ESL supports subtypes and inheritance. Behaviour types can be defined as an extension of an existing behaviour type in which case all the message definitions and export declarations from the parent are included in the child. Whenever an identifier is declared to be of a particular behaviour type, any behaviour that conforms to an extension of the declared type can be supplied.


Behaviour definitions can extend a parent definition in which case the message handlers from the parent are all included in the child (in the same way that Java methods are inherited). Where the message signatures overlap, the child handlers will take precedence, but the parent definitions will be used where none of the child definition match. Parent local definitions are available in the child without qualification using the keyword `super`. This protects the integrity of the parent by making access to parent state and behaviour explicit.

To demonstrate ESL object-oriented features we will develop a Pacman game where inheritance allows position information to be abstracted into a behaviour that is then reused across players and ghosts. Sub-types are used to produce different types of ghosts and players.

The game consists of a maze containing 4 ghosts, some food and a Pacman. The Pacman moves around eating the food and the ghosts move around trying to catch the Pacman. The ghosts can only move horizontally and vertically whilst the Pacman can also move diagonally. No ghost or the Pacman can move through walls. At any given time, the ghosts are in one of two modes: `Scatter` and `Chase`. When in `Scatter` mode, the ghosts try to reach their home corner; each ghost has a different home:



When in `case` mode, the ghosts head towards the Pacman and try to occupy the same location in which case the Pacman loses:



The user interface is defined as an actor that is supplied with messages to update the Pacman maze:

```
1 data GhostInfo = Point(Ghost,Int,Int,Int); // Point(g,type,
2   x,y)
3
4 data GhostState =
5   Scatter
6   | Chase;
7
8 Act GUI {
9   export mouseX:Int; mouseY:Int;
10  State([GhostInfo],Array[Array[Int]]);
11  Player(Int,Int);
12 }
```

The Pacman maze is implemented as an actor with the behaviour type `Maze`. The maze provides the shared world state and ensures that actors move in turn each time the `Move` message is processed:

```
1 Act Maze {
2   export isWall::(Int,Int)→Bool;
3   atHome::(Int,Int)→Bool;
4   getPacman::() → Pacman;
5   getGhosts::() → [Ghost];
6   legalPos::(Int,Int)→Bool;
7   eat::(Int,Int)→Void;
8
9   RegisterGhost(Ghost);
10  RegisterPacman(Pacman);
11  Move;
12 }
```

All the actors in the maze will be sub-types of `Player` which provides basic features for location and starting to play:

```
1 Act Player {
2   export
3     getX::() → Int;
4     getY::() → Int;
5     setX::(Int) → Void;
6     setY::(Int) → Void;
7     getMaze::() → Maze;
8     moveTo::(Int,Int)→Void;
9     move::()→Void;
10    canMove::(Int,Int)→Bool;
11    plan::(Int,
12           (Plan,Plan)→Bool,
13           (Plans)→Plan,
```

```

14     (Plan)→Bool,
15     (Plan)→[Loc],
16     Plans)
17     → Plan;
18 moves::(Plan)→[Loc];
19 deltaOK::(Int,Int)→Bool;
20 initPlans::()→Plans;
21 deltas::[Int];
22
23 Start;
24 }

```

The operation `move` is called to make a single-step move in the game. A player uses `plan` to plan out the move. The arguments of `plan` are as follows:

```
1 plan(max,compare,goal,ok,moves,start)
```

where `max` determines the maximum length of the plan, `compare` is a predicate that compares plans such that `compare(p1,p2)` is true when `p1` is a better plan than `p2`, `goal` maps a collection of plans to a single plan that achieves a goal or returns `null`, `ok` returns true when a plan is legal, `moves` maps a plan to a sequence of locations that are legal next moves; `start` is a collection of starting plans (usually the current position of a player).

The `plan` operation is used to automatically generate next moves for players and is implemented differently for ghosts and the Pacman. Some of the player operations can be implemented and then reused across ghosts and players:

```

1 act player(x:Int,y:Int,maze:Maze)::Player {
2
3   // A basic implementation of the player interface...
4
5   export getX,getY,setX,setY,moveTo,getMaze,move,canMove,
6   moves,plan,deltaOK,initPlans,deltas;
7
8   getX():Int = x;
9   getY():Int = y;
10  setX(v:Int):Void = x := v;
11  setY(v:Int):Void = y := v;
12  moveTo(v1:Int,v2:Int):Void = { x := v1; y := v2; }
13  getMaze():Maze = maze;
14  move():Void = throw[Void]('abstract operation move');
15  canMove(x:Int,y:Int):Bool =
16    maze.legalPos(x,y) or maze.atHome(x,y);
17  deltas:[Int] = [-1,0,1];
18  deltaOK(dx:Int,dy:Int):Bool = not(dx = 0 and dy = 0);
19  initPlans():Plans = [[Loc(self.getX(),self.getY())]];
20
21  getBestPlan(better:(Plan,Plan)→Bool,plans:Plans)::Plan =
22    select1[Plan](plans,null[Plan],fun(p1:Plan)::Bool
23      not(exists[Plan](fun(p2:Plan)::Bool
24        p1 <> p2 and better(p2,p1),plans)));
25
26  plan(max:Int, better:(Plan,Plan)→Bool,
27    goal:(Plans)→Plan,pred:(Plan)→Bool,
28    moves:(Plan)→[Loc],plans:Plans)::Plan =
29    if plans = []
30    then null[Plan];
31    else if max < 0
32    then getBestPlan(better,plans);
33    else
34      let p:Plan = goal(plans);
35      plans:Plans =
36        [p | p ← addMoves(plans,moves), ?pred(p)];
37      in if p = null[Plan]
38        then plan(max-1,better,goal,pred,moves,plans);

```

```

39           else p;
40
41 hasMove(m:Loc)::(Plan)→Bool =
42   fun(p:Plan)::Bool member[Loc](m,p);
43
44 addMoves(plans:Plans,moves:(Plan)→[Loc])::Plans =
45   [ m:p | p ← plans,
46     m ← moves(p),
47     ?(not(exists[Plan](hasMove(m,plans)))) ];
48
49 moves:(Plan) → [Loc] =
50   cache[Plan,Loc,[Loc]](head[Loc],fun(p:Plan)::[Loc]
51     case head[Loc](p) {
52       Loc(x:Int,y:Int) →
53         [ Loc(x+dx,y+dy) |
54           dx ← deltas,
55           dy ← deltas,
56           ?(self.deltaOK(dx,dy)),
57           ?(canMove(x+dx,y+dy)) ];
58     });
59
60 Start → throw[Void]('abstract message Start');
61 }

```

In `player` the implementation of `plan` uses the supplied arguments to map a collection of plans to a single best plan. A plan is a sequence of locations, in reverse order, that are a legal sequence of moves from some starting position. The operation `addMoves` is used by `plan` to add a new move to the end of each plan. A check is made in `addMoves` to ensure that no plan contains a cycle.

In Pacman there are four types of ghost. Each type has a different behaviour which will be implemented as different sub-behaviours of a common parent behaviour `ghost`. The maze is drawn by a Swing-based Java application that will draw the ghosts in different colours, supplied to the Java as integer codes that are the value of `ghostType`:

```
1 Act Ghost extends Player { export ghostType:Int; }
```

We create an empty behaviour type as placeholder in the hierarchy for players:

```
1 Act Pacman extends Player { }
```

Figure 9 defines an abstract ghost behaviour that inherits the behaviour `player`. A ghost exists in one of two states: `Chase` and `Scatter` which controls whether it is trying to catch the Pacman or trying to reach its home position. It achieves this by creating a plan using the local function `plan`. The best plan is that which minimises the distance to the goal location.

Having defined an abstract ghost behaviour we can extend it to produce four different concrete ghost behaviours. In the real game, the ghosts differ by strategy to catch the Pacman, however we will limit the differences to the home locations:

```

1 act ghost1(x:Int,y:Int,m:Maze)::Ghost
2   extends ghost(x,y,1,1,0,m) { }
3 act ghost2(x:Int,y:Int,m:Maze)::Ghost
4   extends ghost(x,y,1,maxHeight-2,1,m) { }
5 act ghost3(x:Int,y:Int,m:Maze)::Ghost
6   extends ghost(x,y,maxWidth-2,1,2,m) { }
7 act ghost4(x:Int,y:Int,m:Maze)::Ghost
8   extends ghost(x,y,maxWidth-2,maxHeight-2,3,m) { }

```

An abstract Pacman is a player that registers itself with the

```

1 act ghost(x:Int,y:Int,homeX:Int,homeY:Int,gType:Int,m:Maze)::Ghost extends player(x,y,m) {
2   export ghostType, move, deltaOK;
3
4   ghostType:Int      = gType;           // The type of the ghost.
5   state::GhostState = Chase;          // Current state, used for planning.
6   changeMode:Int    = 15;             // Change the state after this number of moves.
7   modeCount:Int     = changeMode;     // Current countdown to change state.
8
9   switchMode(s::GhostState)::Void = { state := s; modeCount := changeMode; }
10
11  deltaOK(dx:Int,dy:Int)::Bool = super.deltaOK(dx,dy) and (dx = 0 or dy = 0);
12
13  move()::Void =
14    letrec
15      planDistance(p:Plan,x:Int,y:Int)::Float = case head[Loc](p) { Loc(x0,y0) → distance(x,y,x0,y0); }
16      pred(p:Plan)::Bool = true;
17      better(goalX:Int,goalY:Int)::(Plan,Plan) → Bool = fun(p1:Plan,p2:Plan)::Bool
18        planDistance(p1,goalX,goalY) < planDistance(p2,goalX,goalY);
19      at(goalX:Int,goalY:Int)::(Plans)→Plan = fun(plans:[Plan])::Plan
20        select1[Plan](plans,null[Plan],fun(p:Plan)::Bool head[Loc](p) = Loc(goalX,goalY));
21      getMove(goalX:Int,goalY:Int)::Loc =
22        case self.plan(8,better(goalX,goalY), at(goalX,goalY), pred, self.moves,self.initPlans()) {
23          p when p = null[Plan] → Loc(self.getX(),self.getY());
24          [1]                      → Loc(self.getX(),self.getY());
25          p                         → last[Loc](butlast[Loc](p));
26        }
27    in
28    case state {
29      Scatter when modeCount > 0 →
30        case getMove(homeX,homeY) {
31          Loc(x:Int,y:Int) → { self.moveTo(x,y); modeCount := modeCount - 1; }
32        }
33      Scatter → switchMode(Chase);
34      Chase when modeCount > 0 →
35        case getMove(m.getPacman().getX(),m.getPacman().getY()) {
36          Loc(x:Int,y:Int) → { self.moveTo(x,y); modeCount := modeCount - 1; }
37        }
38      Chase → switchMode(Scatter);
39    }
40  Start → m ← RegisterGhost(self);
41 }

```

Figure 9: Abstract Ghost Behaviour

maze:

```

1 act pacman(x:Int,y:Int,m:Maze)::Pacman
2
3   extends player(x,y,m) {
4
5   Start →
6     m ← RegisterPacman(self);
7 }

```

There will be two different concrete Pacman behaviours. The first allows a human player to control the Pacman using the mouse. This will be achieved by querying the current mouse position:

```

1 act playerPacman(x:Int,y:Int,m:Maze)::Pacman
2
3   extends pacman(x,y,m) {
4
5   export move;
6
7   move()::Void = {
8     let mouseX:Int = gui.mousePosition;
9     mouseY:Int = gui.mousePosition;


```

```

10      x:Int = self.getX(); y:Int = self.getY(); in
11      let dx:Int = mouseX - x;
12      dy:Int = mouseY - y; in
13      let mx:Int = if dx>0 then 1;else if dx<0 then -1;else 0;
14      my:Int = if dy>0 then 1;else if dy<0 then -1;else 0;
15      in {
16        if (mx <> 0) and m.legalPos((x+mx)%maxWidth,y)
17        then self.setX((x+mx)%maxWidth);
18        else
19          if (my <> 0) and
20            m.legalPos(x,y+my)
21            then self.setY(y+my);
22            else
23              if m.legalPos((x+mx)%maxWidth,y+my)
24                then self.moveTo((x+mx)%maxWidth,y+my);
25                else
26                  if x + mx = 0 and m.legalPos(0,y)
27                    then self.setX(maxWidth - 1);
28                    else {}
29                  gui ← Player(x,y);
30                  m.eat(x,y);
31                }
32  }
33 }

```

An alternative implementation of the player provides a machine implemented behaviour. Like ghosts, the automatic movement is defined by a simple planner that tries to maximise the amount of food that is eaten whilst avoiding ghosts. In order to avoid the ghosts, the Pacman must work out the *danger zones* which are those locations that the ghosts can reach:



When the Pacman creates a plan, those plans whose locations stray into danger zones are discarded. The implementation is shown in figure 10.

15. Observers

An ESL application contains a collection of autonomous actors. Unlike traditional applications where there is often a central controller, an actor-based application encourages control to be distributed. This raises the question: how can shared information be used influence the autonomous actors? In ESL, this can be achieved through the use of *observers* which is the ESL implementation of the observer pattern.

Any ESL actor can be observed by one or more actors. Observation occurs by viewing the behaviour of an actor as a state machine. An actor starts in an initial state. Transitions between states occur each time a message is handled by the actor. In most cases, an actor's behaviour involves implementation detail that includes data and messages that are required for housekeeping. Therefore, we distinguish between an internal implementation-specific version of behaviour and an external logical view of behaviour. The logical view, represented as a state machine, is published by any actor that is observable.

Any actor can act as an observer by subscribing to the behaviour that is published by an observable actor. The observer is informed when the observed actors state changes. A single observed actor together with two subscribing observing actors is shown in figure 11.

In order to be observed, an actor's behaviour type must extend the following:

```

5 type Observed[State,Message] = Act {
6   export
7     observeState():() → State;
8     observeMessage::(union { }) → Maybe[Message];
9 }
```

The behaviour type `Observed` has two parameters: `State` and `Message`. The first parameter is the type used to represent the logical states of the observed actor. The second parameter is the type used to represent the logical transitions of the observed actor. An observed actor must define and export two functions: `observeState` that maps implementation states to logical states, and `observeMessage` that maps messages received by the actor to logical transitions.

The type `Maybe` is used to allow some implementation-specific messages received by an observed actor, to remain hidden. The operation `observeMessage` is supplied with an implementation-specific message and must decide whether to return `Nothing` or a value `Something(m)` where `m` is a logical representation of the message.

Note that the domain of `observeMessage` is `union {}` in the definition of `Observed`. This will be replaced with a sub-type of `union {}` in any particular observed behaviour type which must agree with the message interface. The following is a simple example:

```

1 data IntMessage = M1 | M2;
2 data ExtMessage = X1;
3 data ExtState = S(Int);
4
5 Act ObservedExample {
6   export
7     observeState():() → ExtState;
8     observeMessage::(IntMessage) → Maybe[ExtMessage];
9     M1;
10    M2;
11 }
```

Note that the messages defined by `ObservedExample` (`M1` and `M2`) agree with the data type `IntMessage` that is the domain type of `observeMessage`.

The following is an example behaviour that implements the observed type defined above:

```

1 act oBehaviour::ObservedExample {
2   export observeState,observeMessage;
3   count:Int = 0;
4   observeState():ExtState = S(count);
5   observeMessage(m:IntMessage)::ExtMessage =
6     case m {
7       M1 → Something[ExtMessage](X1);
8       M2 → Nothing[ExtMessage];
9     }
10  M1 → count := count + 1;
11  M2 → Count := count - 1;
12 }
```

Note that the definition of `observeMessage` hides occurrences of `M2`.

An observer for a given observed behaviour type must define message handlers for `Start` and `Transition`:

```

1 type Observer[State,Message] = Act {
2   Start(Observed[State,Message],Int,State);
3   Transition(Observed[State,Message],Int,Message,State);
```

```

1 act autoPacman(x:Int,y:Int,m:Maze)::Pacman extends pacman(x,y,m) { export move;
2
3 plan::Plan          = [] ;           // A sequence of locations in reverse order.
4 fleeing::Bool        = false;        // Is the pacman in danger of being eaten?
5
6 moves(p::Plan)::[Loc] = [ Loc(if locX(p) = 0 then maxWidth-1 else locX(p),locY(p)) | p ← super.moves(p) ];
7
8 ghostDistance(l::Loc)::Float = min(1000.0,[ distance(locX(l),locY(l),g.getX(),g.getY()) | g::Ghost ← m.getGhosts() ]);
9
10 dangerous(dangerZones::[[Loc]])(Plan) → Bool = fun(p::Plan)::Bool
11   let canReach::[[Loc]] = take[[Loc]](dangerZones,length[Loc](p)+1);
12   in not(exists[[Loc]](fun(zone::[Loc])::Bool member[Loc](head[Loc](p),zone),canReach));
13
14 nextMove(p::Plan)::Loc = case p { p1 + [1] + p2 when length[Loc](p2) = 1 → l; }
15
16 move():Void =
17   let d::Float = ghostDistance(Loc(self.getX(),self.getY()));
18   in if d < 5.0 and not(fleeing) then { flee(7); fleeing := true; doPlan(); }
19   else if d > 5.0 and fleeing then { eat(9); fleeing := false; doPlan(); }
20   else doPlan();
21
22 ghostMoves::(Loc) → [Loc] = cache[Loc,Loc,[Loc]](id[Loc],fun(l::Loc)::[Loc]
23   case l {
24     Loc(x:Int,y:Int) → [ Loc(x+dx,y+dy) |
25       dx ← super.deltas, dy ← super.deltas, ?(dx=0 or dy=0), ?(not(dx=0 and dy=0)), ?(self.canMove(x+dx,y+dy)) ];
26   });
27
28 getDangerZones(moves::Int)::[[Loc]] =
29   letrec
30     ghostLocs::[Loc] = [ Loc(g.getX(),g.getY()) | g::Ghost ← m.getGhosts() ];
31     zones(zone::[Loc],i:Int,max:Int)::[[Loc]] =
32       if i >= max then []; else let z::[Loc] = [ m | l::Loc ← zone, m::Loc ← ghostMoves(l) ]; in z:zones(z,i+1,max);
33   in zones(ghostLocs,0,moves);
34
35 eat(n:Int):Void =
36   letrec
37     foodCount(plan::Plan)::Int = sum([ food[locY(1)][locX(1)] | l ← plan ]);
38     moreFood(plan1::Plan,plan2::Plan)::Bool = foodCount(plan1) > foodCount(plan2);
39     mostFood(plans::Plans)::Plan = if length[Loc](head[Plan](plans)) < n then null[Plan]; else max[Plan](plans,moreFood);
40   in case self.plan(n,moreFood,mostFood,dangerous(getDangerZones(n+1)),moves,self.initPlans()) {
41     p when p = null[Plan] → plan := [];
42     [1]                   → plan := [];
43     p                     → plan := butlast[Loc](p);
44   }
45
46 flee(n:Int):Void =
47   letrec
48     ok(p1::Plan,p2::Plan)::Bool = true;
49     anything(plans::Plans)::Plan = if length[Loc](head[Plan](plans)) < n then null[Plan]; else head[Plan](plans);
50   in case self.plan(n,ok,anything,dangerous(getDangerZones(n+1)),moves,self.initPlans()) {
51     p when p = null[Plan] → plan := [];
52     [1]                   → plan := [];
53     p                     → plan := butlast[Loc](p);
54   }
55
56 doPlan():Void =
57   if plan = [] then replan();
58   else
59     case plan {
60       p1 + [1] → {
61         plan := p1;
62         case l {
63           Loc(x,y) → { self.moveTo(x,y); gui ← Player(self.getX(),self.getY()); m.eat(self.getX(),self.getY()); }
64         }
65       }
66     }
67
68   replan():Void = if fleeing then flee(7); else eat(9);
69 }
```

Figure 10: Automatic Pacman




Figure 11: Observers

4 }


An actor o_2 whose behaviour implements the type $\text{Observer}[\text{State}, \text{Message}]$ can be registered as an observer with an actor o_1 whose behaviour implements $\text{Observed}[\text{State}, \text{Message}]$ as follows:

1 $\text{o}_1.\text{addObserver}(\text{o}_2)$

Immediately on being added, the observer will receive a message $\text{Start}(\text{o}_1, t, s)$ where t is the current time and s is the logical state of o_1 . Each time a message is processed by o_1 , its operation observeMessage will map to $\text{Maybe}[\text{Message}]$ and if the result is $\text{Something}(m)$ then the observer receives a message transition (o_1, t, m, s) where s is the value produced by $\text{o}_1.\text{observeState}()$ after the message has been handled.

15.1. A Taxi Service

The behaviour of a taxi can be described using the following state machine:



such that the taxi management service can observe the behaviour in order to display its current position and to provide new jobs as soon as the current customer has been dropped-off at their destination. The implementation details of a taxi actor deals with managing the route to the customer and to the customer drop-off. To start, define the type cab as an observable behaviour type:

```

1 data Loc = Loc(Int,Int);
2
3 data Direction = North | East | South | West;
4
5 data CabState =
6   Empty(Loc,Direction)
7 | Full(Loc,Direction);
8
9 data CabMessage =
10  ChangeDir(Direction)
11 | MoveDir(Direction)
12 | Move
13 | Pickup
14 | DropOff
15 | Customer(Int,Int,Int,Int);
16
17 act cab(x0:Int,y0:Int)::Cab {
18   export
19     observeState():()→CabState;
20     observeMessage:(CabMessage) → Maybe[CabMessage];
21   Move;
22   Customer(Int,Int,Int,Int);
23 }
  
```

Note that the type CabMessage is used as both the implementation interface and the logical interface of cab , however the function observeMessage may not return all of the messages in the interface. In fact, it hides the Move and Customer messages:

```

1 act cab(x0:Int,y0:Int)::Cab {
2   export observeState,observeMessage;
3   x:Int          = x0;
4   y:Int          = y0;
5   dir:Direction  = North;
6   plan:[Loc]      = [];
7   isEmpty:Bool   = true;
8   observeState():CabState =
9     if isEmpty
10       then Empty(Loc(x,y),dir);
11     else Full(Loc(x,y),dir);
12   observeMessage(m::CabMessage)::Maybe[CabMessage] =
13     case m {
14       ChangeDir(d) → Something[CabMessage](m);
15       MoveDir(d)   → Something[CabMessage](m);
16       Pickup       → Something[CabMessage](m);
17       DropOff      → Something[CabMessage](m);
18       m            → Nothing[CabMessage];
19     }
20   Customer(xStart:Int,yStart:Int,xEnd:Int,yEnd:Int) → {
21     plan := [Loc(xStart,yStart),Loc(xEnd,yEnd)];
22     self ← Move;
23   }
24   Pickup → isEmpty := false;
25   DropOff → isEmpty := true;
26   Move →
27     case plan {
28       [] → {}
29       Loc(tx,ty):p when tx=x and ty=y and isEmpty → {
30         plan := tail[Loc](plan);
31         self ← Pickup;
32         self ← Move;
33       }
34       Loc(tx,ty):p when tx=x and ty=y and not(isEmpty) → {
35         plan := tail[Loc](plan);
36         self ← DropOff;
37         self ← Move;
38       }
39       Loc(tx,ty):locs when x>tx → self ← MoveDir(West);
40       Loc(tx,ty):locs when x<tx → self ← MoveDir(East);
41       Loc(tx,ty):locs when y>ty → self ← MoveDir(North);
  
```

```

42     Loc(tx,ty):locs when y<ty → self ← MoveDir(South);
43   }
44 MoveDir(North) when dir=North → { y:=y-1; self←Move; }
45 MoveDir(South) when dir=South → { y:=y+1; self←Move; }
46 MoveDir(East) when dir=East → { x:=x+1; self←Move; }
47 MoveDir(West) when dir=West → { x:=x-1; self←Move; }
48 MoveDir(d::Direction) → {
49   self ← ChangeDir(d);
50   self ← MoveDir(d);
51 }
52 ChangeDir(d::Direction) → dir := d;
53 }
```

The type `Uber` is an observer for a cab. It keeps track of the cabs using a table that maps cab actors to a term of type `CabDisplay`:

```

1 Act Uber {
2   Start(Cab,Int,CabState);
3   Transition(Cab,Int,CabMessage,CabState);
4   NewJob;
5 }
6 data CabDisplay = CabDisplay(Int,Int,Direction,Bool);
```

The taxi cab service contains a screen (defined later) that is used to show the current set of taxis and jobs. The Transtion handler monitors the taxis and works out whether they should be allocated a new job. Note that the screen is updated only when a given taxi is updated more than once. This allows all taxis to move before the screen is updated and prevents flicker.

```

1 act uberBehaviour(width:Int,height:Int)::Uber {
2   maxJobs:Int = 8;
3   pending:[Cab] = [];
4   jobs:[Job] = [];
5   customers:[Loc] = [];
6   targets:[Loc] = [];
7   screen:Screen = new screenBehaviour(width,height);
8   display:Hash[Cab,CabDisplay] = new Hash[Cab,CabDisplay];
9   movedCabs:[Cab] = [];
10  checkJobs():Void = {
11    if pending <> [] and jobs <> []
12    then allocateJob();
13    else {}
14  }
15  allocateJob():Void = {
16    case head[Job](jobs) {
17      Job(xStart,yStart,xEnd,yEnd) → {
18        head[Cab](pending)
19        ← Customer(xStart,yStart,xEnd,yEnd);
20        pending := tail[Cab](pending);
21        jobs := tail[Job](jobs);
22      }
23    }
24  }
25 Start(c:Cab,t:Int,s:CabState) → {
26   case s {
27     Empty(Loc(x,y),d) →
28       display.put(c,CabDisplay(x,y,d,false));
29   }
30   screen ← Show(display.vals,customers,targets);
31   pending := pending + [c];
32   checkJobs();
33 }
34 Transition(c:Cab,t:Int,m:CabMessage,s:CabState) → {
35   case s {
36     Empty(Loc(x,y),d) →
37       display.put(c,CabDisplay(x,y,d,false));
38     Full(Loc(x,y),d) →
39       display.put(c,CabDisplay(x,y,d,true));
40   }
```

```

41   case m {
42     Pickup → {
43       case s {
44         Full(Loc(x,y),d) →
45           customers := remove[Loc](Loc(x,y),customers);
46       }
47     }
48     DropOff → {
49       case s {
50         Empty(Loc(x,y),d) → targets :=
51           remove[Loc](Loc(x,y),targets);
52       }
53       pending := pending + [c];
54       checkJobs();
55     }
56     ChangeDir(d) → {}
57     m → {
58       if member[Cab](c,movedCabs)
59       then {
60         movedCabs := [];
61         screen ← Show(display.vals,customers,targets);
62       } else movedCabs := c:movedCabs;
63     }
64   }
65   NewJob when length[Loc](jobs) < maxJobs → {
66     let xStart:Int = random(width);
67     yStart:Int = random(height);
68     xEnd:Int = random(width);
69     yEnd:Int = random(height);
70     in {
71       customers := customers + [Loc(xStart,yStart)];
72       targets := targets + [Loc(xEnd,yEnd)];
73       jobs := jobs + [Job(xStart,yStart,xEnd,yEnd)];
74       checkJobs();
75     }
76   }
77 }
78 NewJob → {}
```

The screen translates taxi, customer and drop-off targets to an HTML table. The message `Show(d,j,t)` contains the current states of the cabs, the waiting customers and their target drop-off locations. The table is created by iterating through positions (x,y) generating cell entries based on the information supplied in the message:

```


1 Act Screen {
2   Show([CabDisplay],[Loc],[Loc]);
3 }
4
5 act screenBehaviour(width:Int,height:Int)::Screen {
6   scale:Int = 38;
7   scaleText:Str = 'font-size:scale'+scaleText+'px';
8   cellScale:Str = (scale + 10) + 'px';
9   frontTaxi:Display = HTML('&#128662;');
10  sideTaxi:Display = HTML('&#128661;');
11  customer:Display = HTML('&#128587;');
12  target:Display = HTML('&#127919;');
13  space:Display = HTML('&nbsp;');
14  tableProps:[Property] = [Prop('border','1'),
15                           Prop('cellpadding','0')];
16  rowProps:[Property] = [Prop('align','center'),
17                         Prop('style',scaleText),
18                         Prop('width',cellScale),
19                         Prop('height',cellScale)];
20 Show(display:[CabDisplay],jobs:[Loc],targets:[Loc]) → {
21   let
22     getDisplay(x:Int,y:Int):Display =
23       let at(d:CabDisplay):Bool =
```

```

24         case d {
25             CabDisplay(dx,dy,d,f) → dx = x and dy = y;
26             d → false;
27         } in
28     let d:CabDisplay =
29         select1[CabDisplay](display,null[CabDisplay],at);
30     in if d = null[CabDisplay]
31         then
32             if member[Loc](Loc(x,y),jobs)
33                 then customer;
34             else
35                 if member[Loc](Loc(x,y),targets)
36                     then target;
37                     else space;
38         else
39             case d {
40                 CabDisplay(x,y,North,b) → frontTaxi;
41                 CabDisplay(x,y,East,b) → sideTaxi;
42                 CabDisplay(x,y,West,b) → sideTaxi;
43                 CabDisplay(x,y,South,b) → frontTaxi;
44             }
45     in edb ← Show('Uber',
46         Table(tableProps,
47             [ Row([], [
48                 [ Data(rowProps, getDisplay(col,row))
49                   | col ← 0..width ]
50                 | row ← 0..height ]]));
51     }
52 }

```

The following screen is generated by the associated `main` behaviour:



```

1 act main::Main {
2
3     noOfCabs:Int = 2;
4     width:Int = 10;
5     height:Int = 10;
6     uber:Uber = new uberBehaviour(width,height);
7     cabs:[Cab] = [ new cab(random(width),random(height))
8                   | n ← 0..noOfCabs ];
9     limit:Int = 40000;
10
11    → {
12        for c in cabs do {
13            c.addObserver(uber);

```

```

14        }
15    }
16
17    Time(n:Int) when n < limit →
18        probably(5) {
19            uber ← NewJob;
20        }
21
22    Time(n:Int) → stopAll();
23
24 }

```

16. EDB

EDB is an editor-based development environment for ESL that integrates syntax checking, type checking, compilation, application execution and run-time debugging. EDB relies on access to a Java compiler as provided by:

```
javax.tools.ToolProvider.getSystemJavaCompiler()
```

At startup, EDB is supplied with the following run-time Java arguments in order:

1. A path to the directory that contains the ESL source code. From the root installation directory, this will be `esl`.
2. A path to a directory containing EDB data that is saved between executions. This includes information about the current state of EDB when it is shut down so that it can be recreated.

EDB consists of a collection of tabs. Each tab has a particular type:

HTML Contains HTML that has been generated by an ESL application or from a file. The HTML is displayed using a browser implemented by JxBrowser¹. Handlers can be registered with EDB that process hyperlinks when they are clicked.

ESL Contains the contents of an ESL file that can be edited within EDB, written to the file system, compiled to produce Java and then run. An ESL tab provides syntax checking and type checking for the ESL language.

Java EDB is not intended to provide extensive support for Java development, however it provides limited support for editing, pretty-printing and compilation of Java source code to support the debugging and modification of Java that is produced when ESL is compiled.

TabbedActor Java based applications that use Swing libraries can be integrated into EDB by defining a class that implements the `edb.editor.TabbedActor` interface and that extends a Java Swing container.

16.1. EDB Interface

The EDB interface is shown in figure 12 with a single ESL tab. The text is highlighted according to ESL syntax rules which

¹<https://www.teamdev.com/jxbrowser>




Figure 12: EDB Showing an ESL Tab

are checked dynamically as the text is modified. Errors are shown in two ways: (1) by underlining the text containing the error in red; (2) as a red marker in the margin. Hovering over either of these with the mouse will describe the error.

The toolbar buttons provide access to EDB functionality. For an ESL tab the following functionality is supported:

button	key	description
		Load file into EDB.
		Create new file.
		Delete file.
		Stop a running application.
		Change font.
	Meta-S	Save file.
		Save file as.
	Meta-T	Touch file and imports.
	Meta-K	Compile ESL to Java.

Pressing command and clicking on an identifier in an ESL editor will jump to its definition. The background menu provides access to all definitions via their name. All the defined names in an ESL file can be copied to the clipboard via the background menu; this is very useful in order to export the names in a file via the `export` clause at the start of the file, which

can be created via copy and paste.

A Java tab provides a subset of the ESL functionality in addition to:

button	key	description
	Meta-J	Pretty print the Java source.
	Meta-R	Compile and run file.

Both types of editor support a background menu that offers functionality for search, find and replace, copy text as a bitmap to the clipboard, and zoom.

16.2. EDB Actor

EDB can be referenced within a running ESL application using the variable `edb` whose type is defined in the module `esl/displays.esl`:

```

1 Act EDB {
2   export
3     math::Math;
4     message::Forall[T] (T->T;
5     display::Forall[T] (Str,T->Void;
6     button::(Str,Str,Str,Str,()>Void)->Void;
7     Show(Str,Display);
8     Filmstrip(Str,[Display]);
9     AddBrowserListener(BrowserListener);
10    Edit(EditType);
11  }
12
13 type Math = {

```

```

14   circlePos:Int,Int,Int,Int) → Point
15 };
16
17 Act BrowserListener {
18   BrowserEvent(Str);
19 }
20
21 data EditType =
22   ESLSource(Str,Str)
23 | JavaSource(Str,Str,Str)
24 | RawText(Str);

```

The EDB object can be used to display graphics (as described below), register hyperlink event listeners, and to edit source code by creating an EDB tab.

16.3. EDB Displays

EDB supports a data type `Display` that can be used to create a variety of graphical displays. Sending the message:

```
1 edb ← Show(label,display);
```

creates (or replaces) an EDB tab with the supplied label and the graphical display. The display type `Display` is defined in appendix B.

16.4. Tables

EDB can display HTML tables, for example the following table:

0	1	2	3	4	5	6	7	8	9
10	11	12	13	14	15	16	17	18	19
20	21	22	23	24	25	26	27	28	29
30	31	32	33	34	35	36	37	38	39
40	41	42	43	44	45	46	47	48	49
50	51	52	53	54	55	56	57	58	59
60	61	62	63	64	65	66	67	68	69
70	71	72	73	74	75	76	77	78	79
80	81	82	83	84	85	86	87	88	89
90	91	92	93	94	95	96	97	98	99

is produced by:

```

1 props::Props = [Prop('style','border: 1px solid black;')];
2
3 table::Display =
4   Table(props,
5     [ Row[],
6       [ Data(props,HTML('' + j))
7         | j ← (i*10)..((i*10)+10)
8       ])
9     | i ← 0..10
10   ]);
11 edb ← Show('mytable',table);

```

Since tables are HTML, it is possible to include links. The links are processed using browser events:

```

1 props::Props = [Prop('style','border: 1px solid black;')];
2
3 table::Display =
4   Table(props,
5     [ Row[],
6       [ Data(props,
7         HTML('<a href="edb:' + j + '">' + j + '</a>'))
8         | j ← (i*10)..((i*10)+10)
9       ])
10      | i ← 0..10
11   ]);
12
13 act tableListener:BrowserListener {
14   BrowserEvent('50') → stopAll();
15   BrowserEvent(s:Str) → print[Str]('you pressed: ' + s);
16 }
17
18 edb ← AddBrowserListener(new tableListener);
19 edb ← Show('mytable',table);

```

Produces the following table:

0	1	2	3	4	5	6	7	8	9
10	11	12	13	14	15	16	17	18	19
20	21	22	23	24	25	26	27	28	29
30	31	32	33	34	35	36	37	38	39
40	41	42	43	44	45	46	47	48	49
50	51	52	53	54	55	56	57	58	59
60	61	62	63	64	65	66	67	68	69
70	71	72	73	74	75	76	77	78	79
80	81	82	83	84	85	86	87	88	89
90	91	92	93	94	95	96	97	98	99

Clicking on any of the links (except 50) prints the number. Clicking on 50 stops the application. Subsequent clicks do nothing.

16.5. Pie Charts


Pie charts can show results. The values associated with each slice are shown as proportions of the pie:

```

1 pie::Display = Pie([], 'Annual Car Sales', 400, 400, [
2   Slice([], 'VW', 20),
3   Slice([], 'Vauxhall', 10),
4   Slice([], 'Audi', 20),
5   Slice([], 'Ford', 10),
6   Slice([], 'Bentley', 3),
7   Slice([], 'Volvo', 40)]);
8
9 edb ← Show('mypie',pie);

```

which produces:



16.6. Line Graphs


Line graphs show data points:

```

1 lgraph::Display =
2   LineGraph([],'Sales','Month','Volume',700,500,[]
3     GLine([],'Product' + i,[]
4       LPoint([],month,random(500)) | month <= 0..12 ])
5     | i <= 0..3 ]);
6
7 edb <- Show('mylines',lgraph);

```

producing:




16.6.1. Supply Chains

Line graphs can be used to present information about a simulation. For example, a supply chain network involves a collection of processing nodes. A job must visit a sequence of processing nodes. When a job is received at a node, it is added to the processing queue. Each time unit, a node can work on a limited number of jobs based on its processing capacity. Jobs stay at a processing node until they are completed and then move on until they have visited all the nodes in their plan, at which point they are completed. We might be interested in experimenting with queue lengths or the number of processing nodes

of a given type that are required in order to prevent delays becoming too large.

Consider the following diagram that shows a job being processed through a number of nested nodes:



The job must be processed by nodes labelled Red, Green and Blue in that order. When the job arrives at a composite node, it must be processed by the nested nodes in the same order. Each leaf node has a queue and a processing capacity. Where a container has more than one child of the same colour, it can choose where to send the job. Without further information, the parent just chooses a child of the right colour at random.

When a job has been processed by a leaf node for the appropriate number of cycles it is sent back to the parent for further processing as defined by its processing path. This can be implemented in ESL and the delays shown using a line graph. The types are defined as follows:

```

1 data Colour =
2   Red
3 | Green
4 | Blue;
5
6 type ShortestTime = Int;
7 type ActualTime = Int;
8 type ProcessTime = Int;
9 type CurrentTime = Int;
10 type QueueLength = Int;
11 type Id = Int;
12 type Stack[T] = [T];
13 type Colours = [Colour];
14 type Jobs = [Job];
15 type Processors = [Processor];

```

The list `colours` define the sequence of nodes that must be visited by each job:

```
1 colours::Colours = [Red,Green,Blue,Blue,Red,Green,Green];
```

In the list above, a job must be processed by a node of the appropriate colour in the order given at each level.

Jobs have the form `Job(i,tc,ts,ta,cs)` where `i` is a job identifier, `tc` is the time left on the current processing node, `ts` is the current aggregate shortest processing time, `ta` is the current actual processing time, and `cs` is a stack of colour paths. The value `ta - ts` defines the delay in job processing for the job. The value `cs` defines the processing that has yet to take place, for example:

```

1 [[Red],[Green,Red]]
15 queue := queue +
16   [setJobShortest(setJobCurrent(j,pTime),pTime)];
17
18 ChildCompleted(j::Job) → {}
19
20 Time(n:Int) → {
21   let processed::Jobs = process(take[Job](queue,qLength));
22     unprocessed::Jobs = drop[Job](queue,qLength); in
23   let completed::Jobs = select[Job](isFinished,processed);
24     incomplete::Jobs = select[Job](incomplete,processed);
25   in {
26     queue := incomplete +
27       [ jobWaits(j) | j ← unprocessed ];
28     for j::Job in completed do {
29       parent ← ChildCompleted(j);
30     }
31   }
32 }
33 }

```

means that the job must be processed on a `Red` node at the current level and then be processed by `Green` then `Red` at the parent level. Each time the job descends to a nested level, the value `cs` is extended at the head by `colours`.

Jobs are defined as a data type as follows together with some useful processing functions:

```

1 data Job = Job(Id,CurrentTime,ShortestTime,ActualTime,Stack
2   [Colours]);
3
4 jobcurrentTime(j::Job)::Int =
5   case j {
6     Job(id,current,shortest,actual,pending) → current;
7   }
8 setJobCurrent(j::Job,current::CurrentTime)::Job =
9   case j {
10    Job(id,n,shortest,actual,pending) →
11      Job(id,current,shortest,actual,pending);
12  }
13 setJobShortest(j::Job,shortest::ShortestTime)::Job =
14   case j { Job(id,n,t,actual,pending) →
15     Job(id,n,shortest+t,actual,pending);
16  }
17 addJobColours(j::Job)::Job =
18   case j { Job(id,n,shortest,actual,pending) →
19     Job(id,n,shortest,actual,colours:pending);
20  }
21 jobWaits(j::Job)::Job =
22   case j {
23     Job(id,n,shortest,actual,pending) →
24       Job(id,n,shortest,actual+1,pending);
25  }

```

Processing nodes are defined as follows, where `Node` is used as the top-most supply-chain network node:

```

1 Act Node {
2   ChildCompleted(Job);
3   Time(Int);
4   Process(Job);
5 }
6
7 Act Processor extends Node {
8   export getColour():Colour;
9 }

```

A `leaf` actor is has a queue that manages a collection of jobs. Each leaf node is capable of performing `qLength` jobs at any time unit, and requires each job to be processed for `pTime` processing units. Once the leaf has completed a job, it will inform its parent:

```

1 act leaf(colour::Colour,
2   parent::Node,
3   pTime::ProcessTime,
4   qLength::QueueLength)::Processor {
5   export getColour;
6   queue::Jobs = [];
7   getColour():Colour = colour;
8   process(jobs::Jobs)::Jobs =
9     [ Job(id,current-1,shortest,actual+1,pending) |
10       Job(id,current,shortest,actual,pending) ← jobs];
11   isFinished(j::Job)::Bool = jobcurrentTime(j) <= 0;
12   incomplete(j::Job)::Bool = not(isFinished(j));
13
14   Process(j::Job) →

```

A container creates some children depending on the level of depth which controls how deeply nested the supply chain network is:

```

1 act container(depth:Int,
2   colour::Colour,
3   parent::Node)::Processor {
4   export getColour;
5
6   getColour():Colour = colour;
7   pTime():Int = 50;
8   qLength():Int = 3;
9   maxChildren():Int = 3;
10  children::Processors = [];
11
12  process(j::Job)::Void =
13    case j {
14      Job(id,n,t0,t1,[]:pending) →
15        parent ← ChildCompleted(Job(id,n,t0,t1,pending));
16        Job(id,n,t0,t1,(c:cs):pending) →
17          let ps::Processors =
18            [ p | p ← children, ?(p.getColour() = c) ];
19            in (nth[Processor](ps,random(length[Processor](ps)))
20             ) ← Process(Job(id,0,t0,t1,cs:pending));
21    }
22
23  → {
24    if depth = 0
25    then children :=
26      [ new leaf(c,self,pTime(),qLength()) |
27        c ← colours,
28        i ← 0..maxChildren() ];
29    else children :=
30      [ new container(depth-1,c,self) |
31        c ← colours,
32        i ← 0..maxChildren() ];
33  }
34  ChildCompleted(j::Job) → process(j);
35
36  Process(j::Job) → process(addJobColours(j));
37
38  Time(n:Int) → {}
39
40 }

```

The ESL application will create and run a supply-chain network and display the results in terms of the number of jobs that have been delayed. The delay statistics are defined as follows:

```

1 minDelay(jobs::Jobs,min::Int)::Int =
2   case jobs {
3     [] → min;
4     Job(id,t,ts,ta,cs):jobs when (ta - ts) < min →
5       minDelay(jobs,actual-possible);
6     j:jobs → minDelay(jobs,min);
7   }
8
9 maxDelay(jobs::Jobs,max::Int)::Int =
10  case jobs {
11    [] → max;
12    Job(id,t,ts,ta,cs):jobs when (ta - ts) > max →
13      maxDelay(jobs,actual-possible);
14    j:jobs → maxDelay(jobs,max);
15  }
16
17 delayedBy(jobs::Jobs,delay::Int)::Int =
18  case jobs {
19    [] → 0;
20    Job(id,t,ts,ta,cs):jobs when (ta - ts) >= delay →
21      1 + delayedBy(jobs,delay);
22    j:jobs → delayedBy(jobs,delay);
23  }

```

The `root` node creates a single container, supplies jobs at a given rate and then displays the results when all the completed jobs are received:

```

1 act root(depth:Int,limit:Int,nodes:Int)::Node {
2
3   child:Processor = new container(depth,Red,self);
4   jobs:Int = 0;
5   completed:Jobs = [];
6   delays():[Int] =
7     minDelay(completed,9999)..maxDelay(completed,0)+1
8
9   ChildCompleted(j:Job) → {
10     completed := j:completed;
11     if length[job](completed) = jobs and now > limit
12     then {
13       print[Str]('completed ' + jobs + ' jobs');
14       let
15         graph:Display =
16           LineGraph([],'
17             Processing ' + jobs + ' Jobs in ' +
18               (now/1000) +
19                 ' seconds',
20               'Delay',
21               'Number of Jobs Delayed by at Least',
22               1200,500,[
23                 GLine([],'Jobs',[

24                   LPoint([],delay,delayedBy(completed,delay)) |
25                     delay ← delays(),
26                     ?(delay > 0),
27                     ?(delay % 10 = 0),
28                     ?(delayedBy(completed,delay) > 0)]);
29
30       in {
31         edb ← Show('Supply Chain Results',graph);
32         stopAll();
33       }
34     } else {}
35
36   Process(j:Job) → child ← Process(j);
37
38   Time(n:Int) when n < limit → {
39     for i in 0..nodes do {
40       jobs := jobs + 1;
41       self ← Process(Job(jobs,0,0,0,[]));
42     }
43   }

```

```

44
45   Time(n:Int) → {}
46 }

```


Given the following:

```
1 rootNode::Node = new root(2,5000,30);
```

produces the results shown in figure 13.

16.7. Pictures

EDB provides a simple model of picture elements that map directly onto SVG. The following picture:



My Picture

is produced by:


```

1 picture::Display = Picture(500,500,[

2   Rectangle(100,100,300,300,
3     'fill:rgb(0,0,255);stroke-width:3;stroke:rgb(0,0,0)'),
4   Circle(250,250,100,
5     'fill:rgb(0,255,0);stroke-width:3;stroke:rgb(0,0,0)'),
6   Line(200,250,300,250,
7     'stroke-width:3;stroke:rgb(255,0,0)'),
8   Image(110,110,50,50,
9     'https://img.icons8.com/ios/1600/clock.png'),
10  Text(110,380,'My Picture',
11    'stroke:rgb(255,255,255)')
12 ]);
13
14 edb ← Show('mypicture',picture);

```

The following example shows how pictures can be combined with filmstrips to produce an animation. A junction is controlled by traffic lights:



Red cars arrive on the right and green cars on the left. The junction is controlled by traffic lights. Red cars arrive more frequently, and a monitor is used to detect when the queues at the junction are building up so that queues are managed effectively by changing the lights:





Figure 13: Supply Chain Delays



The ESL junction simulation runs and captures the state of the junction as a sequence of actions. The actions are then transformed into a sequence of pictures that can be displayed as a filmstrip. The implementation of the simulation is as follows:

```

1 Act Main { Time(Int); }
2
3 Act TrafficLights {
4   export left::TrafficLight; right::TrafficLight;
5 }
6
7 Act TrafficLight {
8   export colour::Str; change::()→Void;
9 }
10
11 Act Approach {
12   export getQueue::()→Int; prioritise::()→Bool;
13   move::()→Void; Time(Int);
14 }
15
16 Act Monitor { Time(Int); }
17
18 data Action =
19   MoveLeft
20 | MoveRight
21 | ArriveLeft
22 | ArriveRight;
23
24 data MonitorState = PriorityRight | Even | Stuck;
```

Traffic lights are defined as passive behaviours that will be changed by the junction monitor:

```

1 act light(c::Str)::TrafficLight {
2   export colour, change;
3   colour::Str = c;
4   change()::Void =
```

```

5   case colour {
6     'RED' → colour := 'GREEN';
7     'GREEN' → colour := 'RED';
8   }
9 }
10
11 act trafficLights(l::TrafficLight,
12   r::TrafficLight)::TrafficLights {
13   export left,right;
14   left::TrafficLight = l;
15   right::TrafficLight = r;
16 }
```

There are two approaches at the junction. Each approach is implemented as a separate actor with a traffic light. The approach has a queue that is a sequence of integers. The head integer defines the current number of cars waiting at the approach. When a car arrives or leaves, the head is duplicated and changed so that the queue contains a historic record that can be used by the monitor to calculate the build-up of cars. When anything changes, the action is saved on a list of actions that will be used to create pictures:

```

1 act approach(id::Str,
2   light::TrafficLight,
3   probOfNewCar::Int)::Approach {
4
5   export getQueue,move,prioritise;
6
7   queue::[Int] = [0]; // A historical record of the cars
8   at the approach.
9   getQueue()::Int =
10   grab(queue) {
11     head[Int](queue);
12   }
13
14   prioritise()::Bool =
15   grab(queue) {
16     if length[Int](queue) > qtLimit
17     then forall[Int](fun(n:Int)::Bool
18       n > qLimit,take[Int](queue,qtLimit));
```

```

19     else false;
20 }
21
22 move():Void =
23   grab(queue,actions) {
24     if light.colour = 'GREEN'
25     then
26       if head[Int](queue) > 0
27       then {
28         queue := (head[Int](queue) - 1) : queue;
29         actions := actions +
30           [if id='left' then MoveLeft else MoveRight];
31       } else {}
32     else {}
33   }
34
35 Time(n:Int) → {
36   probably(probOfNewCar) {
37     grab(queue,actions) {
38       queue := (head[Int](queue) + 1) : queue;
39       actions := actions +
40         [if id='left' then ArriveLeft else ArriveRight];
41     }
42   }
43   move();
44 }
45 }
```

The junction monitor controls the traffic lights to try to minimise the length of the queues. Where there is a conflict the monitor prioritises the right hand side:

```

1 act mon(left::Approach,
2         right::Approach,
3         lights::TrafficLights)::Monitor {
4
5   queueRight():Bool = right.getQueue() > 0;
6
7   allowRight():Void =
8     if lights.right.colour = 'RED'
9     then {
10       lights.right.change();
11       lights.left.change();
12     } else {}
13
14   allowLeft():Void =
15     if lights.left.colour = 'RED'
16     then {
17       lights.left.change();
18       lights.right.change();
19     } else {}
20
21 Time(n:Int) →
22   case monitorState {
23     PriorityRight →
24       if right.prioritise()
25       then allowRight();
26     else
27       if left.prioritise()
28       then allowLeft();
29     else
30       if queueRight()
31       then allowRight();
32     else allowLeft();
33   Even → probably(50) allowLeft(); else allowRight();
34 }
35 }
```

The junction is created as a collection of elements:

```

2 l1::TrafficLight      = new light('GREEN');
3 left::Approach        = new approach('left',l1,20);
4 right::Approach       = new approach('right',l2,30);
5 lights::TrafficLights = new trafficLights(l1,l2);
6 m:Monitor              = new mon(left,right,lights);
```

After running the junction simulation for a given length of time, actions contains a record of all the steps taken. Each action can be transformed into a picture that shows the state of the junction at that instant:

```

1 edb ← Filmstrip('Traffic Filmstrip',actionsToPictures
                  (0,0,0,actions));
```

where the function 'codeactionsToPictures is defined in figure ??.

16.8. Graphs

EDB can display graphs using GraphViz. For example the following command:


```

1 graph::Display = Graph([], [
2   Node([], 0, HTML('0')),
3   Node([], 1, HTML('1')),
4   Node([], 2, HTML('2')),
5   Node([], 3, HTML('3'))], [
6   Edge([], 0, 1, HTML('e1')),
7   Edge([], 0, 2, HTML('e2')),
8   Edge([], 1, 3, HTML('e3')),
9   Edge([], 2, 3, HTML('e4'))]);
```

```

10
11 edb ← Show('mygraph',graph);
```

produces the following display:



Properties in graphs, nodes and edges are used to control GraphViz. Figure 14 shows an example of EDB displays being used to create planar graphs and then colour the nodes. Any planar graph can be coloured using no more than 5 colours using Kempe's algorithm. The node properties are used to set the fill colour. The following is an example output:

```

1 l1::TrafficLight      = new light('RED');
```

```

1 leftColour::Str      = 'fill:green;stroke:green;'; rightColour::Str      = 'fill:red;stroke:red;';
2 dHd::([Display]) → Display = head[Display]; dTl::([Display]) → [Display] = tail[Display];
3
4 carSize:Int          = 30;           carGap:Int          = 2;           leftCarY:Int      = (border+(width/4));
5 maxCarsWaiting:Int   = 7;            width:Int           = 4*carSize;        rightCarY:Int      = (border+width-(width/4));
6 border:Int            = 50;           waits:Int           = (carSize+carGap)*maxCarsWaiting; pWidth:Int       = 1500;
7 neckLength:Int       = 100;          neckWidth:Int      = carSize+10;      pHeight:Int      = 1000;
8 gapLength:Int         = 150;          gapWidth:Int       = 100;           black::Str       = 'stroke:black;';
9
10 leftTop::PictureElement = Line(border,border,border+waits,border,black);
11 leftBottom::PictureElement = Line(border,border+width,border+waits,border+width,black);
12 rightTop::PictureElement = Line(border+waits+gapLength,border,border+waits+gapLength+waits,border,black);
13 rightBottom::PictureElement = Line(border+waits+gapLength,width+border,border+waits+gapLength+waits,width+border,black);
14 neck:[PictureElement] = topNeck+bottomNeck;
15 centre:[PictureElement] = Line(border,border+width/2,border+(waits*2)+gapLength,border+width/2,'stroke:grey;stroke-dasharray:5,5;');
16 road:[PictureElement] = [leftTop,leftBottom]+neck+[rightTop,rightBottom]+[centre];
17
18 topNeck:[PictureElement] = [
19   Line(border+waits,border,border+waits+(gapLength-neckLength)/2,border+(width-neckWidth)/2,black),
20   Line(border+waits+(gapLength-neckLength)/2,border+(width-neckWidth)/2,border+waits+gapLength-(gapLength-neckLength)/2,border+(width-neckWidth)/2,black),
21   Line(border+waits+gapLength-(gapLength-neckLength)/2,border+(width-neckWidth)/2,border+waits+gapLength,border,black)];
22
23 bottomNeck:[PictureElement] = [
24   Line(border+waits,border+width,border+waits+(gapLength-neckLength)/2,border+width-(width-neckWidth)/2,black),
25   Line(border+waits+(gapLength-neckLength)/2,width+border-(width-neckWidth)/2,border+waits+gapLength-(gapLength-neckLength)/2,width+border-(width-neckWidth)/2,
         black),
26   Line(border+waits+gapLength-(gapLength-neckLength)/2,width+border-(width-neckWidth)/2,border+waits+gapLength,border+width,black)];
27
28 addCarOnRight(pos:Int,label:Int,d:Display)::Display = case d { Picture(x,y,ps) → Picture(x,y,carOnRight(pos,label)+ps); }
29 addCarOnLeft(pos:Int,label:Int,d:Display)::Display = case d { Picture(x,y,ps) → Picture(x,y,carOnLeft(pos,label)+ps); }
30 addCarAtNeckLeft(label:Int,d:Display)::Display = case d { Picture(x,y,es) → Picture(x,y,neckCarLeft(label)+es); }
31 addCarAtNeckRight(label:Int,d:Display)::Display = case d { Picture(x,y,es) → Picture(x,y,neckCarRight(label)+es); }
32
33 actionsToPictures(qLeft:Int,qRight:Int,startL:Int,startR:Int,as:[Action]):[Display] =
34 case as {
35   [] → [];
36   ArriveLeft:as → picture(qLeft+1,qRight,startL,startR):actionsToPictures(qLeft+1,qRight,startL,startR,as);
37   ArriveRight:as → picture(qLeft,qRight+1,startL,startR):actionsToPictures(qLeft,qRight+1,startL,startR,as);
38   MoveLeft:as → movesRight(startL+1,picture(qLeft-1,qRight,startL+1,startR):actionsToPictures(qLeft-1,qRight,startL+1,startR,as));
39   MoveRight:as → movesLeft(startR+1,picture(qLeft,qRight-1,startL,startR+1):actionsToPictures(qLeft,qRight-1,startL,startR+1,as));
40 }
41
42 movesRight(label:Int,ps:[Display]):[Display] = addCarAtNeckLeft(label,dHd(ps)):exitRight(0,label,dTl(ps));
43 movesLeft(label:Int,ps:[Display]):[Display] = addCarAtNeckRight(label,dHd(ps)):exitLeft(0,label,dTl(ps));
44
45 exitRight(pos:Int,label:Int,ds:[Display]):[Display] =
46   if (pos = maxCarsWaiting) or (ds = []) then ds; else addCarOnRight(pos,label,dHd(ds)):exitRight(pos+1,label,dTl(ds));
47 exitLeft(pos:Int,label:Int,ds:[Display]):[Display] =
48   if (pos = maxCarsWaiting) or (ds = []) then ds; else addCarOnLeft(pos,label,dHd(ds)):exitLeft(pos+1,label,dTl(ds));
49
50 carOnRight(pos:Int,label:Int):[PictureElement] =
51   Circle(border+waits+gapLength+carSize+(carSize*pos),border+width/4,carSize/2,leftColour),
52   Text(border+waits+gapLength+carSize+(carSize*pos)-5,5+border+width/4,label+',');
53
54 carOnLeft(pos:Int,label:Int):[PictureElement] =
55   Circle(border+waits-(carSize+(carSize*pos)),border+(3*(width/4)),carSize/2,rightColour),
56   Text(border+waits-((carSize+(carSize*pos))+5),5+border+(3*(width/4)),label+',');
57
58 neckCarLeft(label:Int):[PictureElement] =
59   Circle(border+waits+(gapLength/2),border+width/2,carSize/2,leftColour),
60   Text(border+waits+(gapLength/2)-5,5+border+width/2,''+label,'');
61
62 neckCarRight(label:Int):[PictureElement] =
63   Circle(border+waits+(gapLength/2),border+width/2,carSize/2,rightColour),
64   Text(border+waits+(gapLength/2)-5,5+border+width/2,''+label,'');
65
66 qLeftPicture(cars:Int,startL:Int):[PictureElement] = if cars = 0 then []; else leftQCarPicture(cars-1,startL+cars)+qLeftPicture(cars-1,startL);
67 qRightPicture(cars:Int,startRight:Int):[PictureElement] = if cars = 0 then []; else rightQCarPicture(cars-1,startRight+cars) + qRightPicture(cars-1,startRight);
68
69 leftQCarPicture(position:Int,label:Int):[PictureElement] =
70   let x:Int = border+waits-((carSize+carGap)*position)+carSize;
71   let y:Int = leftCarY;
72   in [Circle(x,y,carSize/2,leftColour), Text(x-5,y+5,''+label,'')];
73
74 rightQCarPicture(position:Int,label:Int):[PictureElement] =
75   let x:Int = border+waits+gapLength+((carSize+carGap)*position)+carSize;
76   let y:Int = rightCarY;
77   in [Circle(x,y,carSize/2,rightColour), Text(x-5,y+5,''+label,'')];
78
79 picture(qLeft:Int,qRight:Int,startL:Int,startR:Int):Display = Picture(pWidth,pHeight,road+qLeftPicture(qLeft,startL) + qRightPicture(qRight,startR));


```

```

1 export main;
2
3 import 'esl/lists.esl', 'esl/displays.esl';
4
5 nodeColour(n::Node)::Str = propValue(nodeProps(n), 'fillcolor');
6 setNodeColour(n::Node,c::Str)::Node = setNodeProps(n,setProperty(nodeProps(n), 'fillcolor', c));
7 edgeBetween(e::Edge,n1::Node,n2::Node)::Bool = edgeTo(e,n1,n2) or edgeTo(e,n2,n1);
8 edgeTo(e::Edge,n1::Node,n2::Node)::Bool = edgeSource(e) = nodeId(n1) and edgeTarget(e) = nodeId(n2);
9 edgeOn(e::Edge,n::Node)::Bool = edgeSource(e) = nodeId(n) or edgeTarget(e) = nodeId(n);
10
11 kempe(graph::Display,colours::[Str])::Display =
12 // Simple algorithm to colour a planar graph.
13 // This is taken from http://www.cs.princeton.edu/~appel/Color.pdf
14 // graph must be planar and has no more than 3v - 6 edges.
15 case graph {
16 Graph(props,[],edges) → graph;
17 Graph(props,nodes,edges) →
18 let degreeLessThan5(n::Node)::Bool = length[Edge]([ e | e ← edges, ?(edgeOn(e,n)) ]) <= 5; in
19 let n::Node = select1[Node](nodes,null[Node],degreeLessThan5); in
20 let es::[Edge] = [ e | e ← edges, ?(edgeSource(e) = nodeId(n) or edgeTarget(e) = nodeId(n)) ]; in
21 let g::Display = kempe(Graph(props,remove[Node](n,nodes),removeAll[Edge](es,edges)),colours);
22 in case g {
23 Graph(props,nodes,edges) →
24 let adjacent::[Node] = [ n0 | n0 ← nodes, ?(exists[Edge](fun(e::Edge)::Bool edgeBetween(e,n,n0),es)) ]; in
25 let usedColours::[Str] = [ nodeColour(n) | n ← adjacent ]; in
26 let available::[Str] = removeAll[Str](usedColours,colours);
27 in Graph(props,setNodeColour(n,head[Str](available)):nodes,edges+es);
28 }
29 }
30
31 mkPlanar(v:Int)::Display = Graph([], [
32 [ Node([Prop('style','filled'),Prop('fillcolor',''))],i,HTML('N' + i)) | i ← 0..v ],
33 [ Edge([],source,target,HTML('')) |
34 i ← 0..random((3*v)-6),
35 source ← [random(v)],
36 target ← [random(v)],
37 ?(source <> target) ];
38
39 act main:Act {} {
40 → edb ← Show('graph',kempe(mkPlanar(20),['red','green','blue','gold','deeppink']));
41 }

```

Figure 14: Graph Colouring



16.9. Filmstrips

An EDB *filmstrip* is a list of displays that are shown on an EDB tab with a slider. When the slider is moved, the tab is updated to show the display at the appropriate index in the list. The following shows a sequence of line graphs where the slider can be moved to a month causing the graph to show the sales up to and including that month:





```

7   Sale(4,120),
8   Sale(5,200),
9   Sale(6,230),
10  Sale(7,250),
11  Sale(8,300),
12  Sale(9,350),
13  Sale(10,400),
14  Sale(11,340),
15  Sale(12,460)];
16
17
18 lgraphs::[Display] = [
19  LineGraph([],'Sales','Month','Volume',400,200,[  

20    GLine([],'Product',[ LPoint([],m,s)  

21      | Sale(m,s) ← take[Sale](salesFigures,month)
22    ])
23  ]) | month ← 0..13 ];
24
25edb ← Filmstrip('sales',lgraphs);

```



16.10. Sequence Diagrams

Sequence diagrams are a useful way of visualising the dynamic progress of an application. For example, sequential factorial is usually defined recursively so that each calculation of $!n$ waits on the stack until the calculation of $!(n - 1)$ returns. Concurrent factorial can use actors to implement that stack frames that are waiting for results and allow any number of factorial calculations to occur at the same time. This is done as follows:

```

1 type Customer = Act { Value(Int,Int,Int); }
2 type Fact = Act{ Get(Int,Int,Int,Customer); }
3
4 act fact::Fact {
5   Get(0,c) → {
6     c ← Value(1);
7   }
8   Get(n,c) →
9     let cc::Customer = new cust(n,c);
10    in self ← Get(n-1,cc);
11  }
12
13 act cust(n:Int,c:Customer)::Customer {
14   Value(m) → c ← Value(n*m);
15 }
16
17 act main::Customer {
18   f::Fact = new fact;
19   → {
20     fact ← Get(3,self);
21     fact ← Get(3,self);
22   }
23   Value(n) → print[Str]('result = ' + n);
24 }

```

When `fact` is send a request to calculate a factorial, it tests whether the supplied value is 0. If so, it immediately replies to the customer `c` supplying 1. Otherwise, the supplied value is `n` and the factorial actor sends itself a message to calculate $!n-1$ and to reply to a newly created customer.


Although this is recursive, it does not prevent interleaving of factorial requests. This can be seen if we create a sequence diagram for two different factorial calculations:

The filmstrip is generated as follows:

```

1 data Sale = Sale(Int,Int);
2
3 salesFigures::[Sale] = [
4   Sale(1,100),
5   Sale(2,110),
6   Sale(3,150),

```



The diagram is produced by creating an EDB sequence diagram. Each factorial documents its progress as a sequence of messages tagged with an identifier that designates its thread of execution (shown as [0] and [1] on the diagram). Once the factorials are completed, a value of the form Sequence(actors, messages) is created and displayed where actors contains a representation of all the actors involved in the calculation and messages is a list of messages sent between them. The ESL code is shown below:

```

1 type Customer = Act { export id:Int; Value(Int,Int,Int); }
2 type Fact = Act{ export id:Int; Get(Int,Int,Int,Customer); }
3
4 data M =
5   GetValue(Int,Int,Int,Fact,Fact,Customer,Customer)
6 | ReturnValue(Int,Int,Int,Customer,Customer)
7 | Zero(Int,Int,Fact,Customer)
8 | Start(Int,Int,Customer,Fact);
9
10 actors:[Actor] = [];
11 messages:[M] = [];
12
13 addActor(behaviour:Str):Int =
14   grab(actors) {
15     let id:Int = length[Actor](actors);
16     in {
17       actors := actors + [Actor(id,0,timeOut,behaviour)];
18       id;
19     }
20   }
21
22 act fact:Fact {
23   export id;
24   id:Int = 0;
25   → id := addActor('fact');
26   Get(0,thread,time:Int,c:Customer) → grab(messages) {
27     messages := messages + [Zero(thread,time,self,c)];
28     c ← Value(thread,time+1,1);
29   }
30   Get(n:Int,thread,time:Int,c) → {
31     let cc:Customer = new cust(n,c);
32     in grab(messages) {
33       messages := messages +
34         [GetValue(n-1,thread,time,self,self,c,cc)];
35       self ← Get(n-1,thread,time+1,cc);
36     }
37   }
38 }
39
40 act cust(n:Int,c:Customer):Customer {
41   export id;
42   id:Int = 0;
43   → id := addActor('cust');
44   Value(thread,time:Int,m:Int) → grab(messages) {
45     messages := messages +
46       [ReturnValue(n*m,thread,time,self,c)];
47
48   setDeath(id,time);
49   c ← Value(thread,time+1,n*m);
50 }
51
52 translate(m:M)::Message =
53   case m {
54     [] → [];
55     Start(n,id,c,f) →
56       Message(1,c.id,f.id,['+id+]Start('+n+'));
57       GetValue(n,id,t,src,tgt,srcC,tgtC) →
58         Message(t,src.id,tgt.id,
59           '+id+]Get('+n+', '+srcC.id+', '+tgtC.id+'));
60       ReturnValue(n,id,t,src,tgt) →
61         Message(t,src,tgt.id,['+id+]Return(' + n + ')');
62       Zero(id,t,f,c) →
63         Message(t,f.id,c.id,['+id+]One');
64   }
65
66 getActor(id:Int,as:[Actor]):Actor =
67   case as {
68     Actor(i,time,death,b):l when i = id →
69       Actor(i,time,death,b);
70     a:l → getActor(id,l);
71   }
72
73 changeDeath(a:Actor,t:Int):Actor =
74   case a {
75     Actor(i,birth,death,b) → Actor(i,birth,t,b);
76   }
77
78 setDeath(id:Int,time:Int):Void =
79   let oldActor:Actor = getActor(id,actors); in
80   let newActor:Actor = changeDeath(oldActor,time);
81   in grab(actors) {
82     actors := subst[Actor](newActor,oldActor,actors);
83   }
84
85 act main:Customer {
86   export id;
87   id:Int = 0;
88   counter:Int = 0;
89   count:Int = 1;
90   f:Fact = new fact;
91   numberOffacts:Int = 2;
92   computeFact(n:Int):Void = grab(messages) {
93     messages := messages + [Start(n,counter,self,f)];
94     counter := counter + 1;
95     f ← Get(n,counter-1,1,self);
96   }
97   → {
98     id := addActor('Main');
99     computeFact(3);
100    computeFact(3);
101  }
102  Value(threadId:Int,time:Int,n:Int) → {
103    setDeath(id,time);
104    if count = numberOffacts
105    then {
106      edb ← Show('Factorial',Sequence(actors,[translate(m) | m ← messages]));
107      stopAll();
108    } else count := count + 1;
109  }
110 }
111
112 }
```

16.11. Combining Pictures

The Display data type allows pictures to be combined using trees. A tree consists of a collection of h-boxes and v-boxes

that EDB will auto-layout. These are similar to HTML tables, but the size of the tree can be controlled. The following example shows how a filmstrip of trees can be constructed to create a simple predator-prey application. The predator-prey world consists of marked locations:

```
1 data Location =
2   EmptyLoc
3 | PredLoc
4 | PreyLoc
5 | Rock;
```

Predators and prey, are located at points in the world and can be requested to move:

```
1 Act Predator {
2   export getX():Int;
3   getY():Int;
4   at:(Int,Int)→Bool;
5   getId():Int;
6   Move;
7 }
8 Act Prey {
9   export getX():Int;
10  getY():Int;
11  at:(Int,Int)→Bool;
12  Move;
13 }
```

The board is a nested list of locations that is set up at random:

```
1 data Point = Point(Int,Int);
2 pointX(p:Point):Int = case p { Point(x,y) → x; }
3 pointY(p:Point):Int = case p { Point(x,y) → y; }
4 board:=[[Location]] = [];
5 rocks:=[Point] = drop[Point](points,numOfPredators+1);
6 points:=[Point] =
7 letrec generate(ps:=[Point],n:Int):=[Point] =
8   if n = 0
9   then []
10  else
11    let x:Int = random(width);
12    y:Int = random(height);
13    in if member[Point](Point(x,y),ps)
14    then generate(ps,n);
15    else Point(x,y):generate(Point(x,y):ps,n-1);
16 in generate([],numOfPredators+1+numOfRocks);
```

Control of positions uses the following predicates:

```
1 onBoard(x:Int,y:Int):Bool =
2   (x >= 0) and (x < width) and (y >= 0) and (y < height);
3 onRock(x:Int,y:Int):Bool =
4   member[Point](Point(x,y),rocks);
5 legalPreyPos(x:Int,y:Int):Bool =
6   (not(onRock(x,y))) and onBoard(x,y) and
7   not(exists[Predator](fun(p:Predator):Bool
8     p.at(x,y),predators));
9 legalPredatorPos(x:Int,y:Int):Bool =
10  (not(onRock(x,y))) and onBoard(x,y) and
11  (not(exists[Predator](fun(p:Predator):Bool
12    p.at(x,y),predators))) and
13  not(thePrey.at(x,y));
```

When they make a move, each of the actors in the world save their activity as a message on a global list. This will be used to create a graphical display of each snapshot of the world and eventually turned into a filmstrip:

```
1 data Message =
```

```
2   MovePredator(Int,Int,Int)
3 | MovePrey(Int,Int);
4 messages:[Message] = [];
```

A predator moves towards the prey. The variable turn is used to achieve equitable movement:

```
1 act predator(id:Int,x:Int,y:Int):Predator {
2   export getX, getY, at, getId;
3   getX():Int = x;
4   getY():Int = y;
5   getId():Int = id;
6   at(px:Int,py:Int):Bool = (px = x) and (py = y);
7
8   move(dx:Int,dy:Int):Void = {
9     if legalPredatorPos(x+dx,y+dy)
10    then {
11      x := x + dx;
12      y := y + dy;
13      messages := messages + [MovePredator(id,x,y)];
14    } else { }
15  }
16
17 Move → grab(messages) {
18   if turn = id
19   then {
20     turn := (turn + 1) % (numOfPredators+1);
21     let dx:Int =
22       if thePrey.getX() > x
23       then 1;
24       else if thePrey.getX() < x
25       then -1;
26       else 0;
27     dy:Int =
28       if thePrey.getY() > y
29       then 1;
30       else if thePrey.getY() < y
31       then -1;
32       else 0;
33     in {
34       if (dy >> 0) and legalPredatorPos(x,y+dy)
35       then move(0,dy);
36       else if (dx >> 0) and legalPredatorPos(x+dx,y)
37       then move(dx,0);
38       else probably(50) {
39         move(0,1-random(3));
40       } else move(1-random(3),0);
41     }
42   } else {}
43 }
44 }
```

The predators are created in a list:

```
1 predators:[Predator] = [
2   new predator(p,pointX(nth[Point](points,p)),pointY(nth[
3   Point](points,p)))
4 | p:Int ← 0..numOfPredators
4 ];
```

The prey moves away from the predators. Unlike the predators, the prey can move diagonally:

```
1 act prey(x:Int,y:Int):Prey {
2
3   export getX, getY, at;
4
5   getX():Int = x;
6   getY():Int = y;
7
8   at(px:Int,py:Int):Bool = (px = x) and (py = y);
```

```

9   dir(n:Int)::Int      = if n < 0 then -1; else 1;
10  dirX:Int = 1 - random(3);
11  dirY:Int = 1 - random(3);
12
13 move(dx:Int,dy:Int)::Void = grab(messages) {
14   if legalPreyPos(x+dx,y+dy)
15     then {
16       x := x + dx;
17       y := y + dy;
18       messages := messages + [MovePrey(x,y)];
19     } else {}
20   }
21
22 predatorWins()::Void = {
23   print[Str]('predator wins!');
24   stop := true;
25 }
26
27 legalMoves()::[Point] = [ Point(dx,dy) |
28   dx ∈ [-1,0,1],
29   dy ∈ [-1,0,1],
30   ?(dx > 0 or dy > 0),
31   ?legalPreyPos(x+dx,y+dy)
32 ]
33
34
35 changeDir()::Void =
36   case legalMoves() {
37     Point(dx,dy):ps → { dirX := dx; dirY := dy; }
38     [] → {}
39   }
40
41 Move → grab(messages) {
42   if turn = numOfPredators
43     then {
44       turn := (turn + 1) % (numOfPredators+1);
45       if legalMoves() = []
46         then predatorWins();
47       else if legalPreyPos(x+dirX,y+dirY)
48         then move(dirX,dirY);
49       else changeDir();
50     } else {}
51   }
52 }

```

A single prey is created:

```

1 thePrey::Prey =
2   let x:Int = pointX(nth[Point](points,numOfPredators));
3   y:Int = pointY(nth[Point](points,numOfPredators));
4   in new prey(x,y);

```


At this point we have sufficient to run the world and send all actors a `Move` message. A sequence of messages can be transformed into a snapshot where the occupied locations. The snapshots can be merged and the resulting composite snapshots merged using `mapBoard`:



```

1 mapMessage(m::Message)::[[Location]] =
2   case m {
3     MovePredator(i,x0,y0) →
4       [[ if (x=x0) and (y=y0)
5         then PredLoc
6         else if onRock(x,y)
7           then Rock
8         else EmptyLoc
9         | x ← 0..width ]
10        | y ← 0..height ];
11     MovePrey(x0,y0) →
12       [[ if (x=x0) and (y=y0)

```

The resulting list of displays is then presented as a filmstrip. The following shows steps in an example filmstrip:





17. Excel Interface

ESL applications that are based on large data sets can read and write to Excel spreadsheets. The spreadsheet is read into an ESL data representation, can then be used as data in an ESL application, and if appropriate the data can be modified and written back out to a spreadsheet. The data representation for an spreadsheet is:

```

1 data ExcelCell =
2   XStr(Str)
3 | XFloat(Float)
4 | XBool(Bool)
5 | XDate(Str)
6 | XFormula(Str)
7 | XBlank
8 | XUnknown;
9
10 type ExcelColumn = [ExcelCell];
11 data ExcelRow    = XRow([ExcelCell]);
12 data ExcelSheet = XSheet(Str,[ExcelRow]);
13 data Excel      =
14   XWorkbook([ExcelSheet])
15 | ExcelError(Str);

```

A workbook is a list of named data sheets. Each sheet is a list of rows containing lists of cells. A cell is a value tagged with its type. The Excel interface supports strings, numbers, booleans, dates and formulas. Note that all numbers will be translated to floats. The format of a date is that produced by the following Java date format:

```
1 "dd/mm/yyyy"
```


A workbook is read from the file system using the following function defined in `esl/excel.esl`:

```
1 readWorkbook::(Str) → Excel
```

and is written to the filesystem using:

```
1 writeWorkbook::(Str,Excel) → Excel
```

Consider the following simple spreadsheet:




	A	B	C	D	E	F
1	Model	Price	Sales	Discontinued	Date	
2	Racer	1000.5	53	TRUE	05/01/2019	
3	Pro-Racer	2000.25	22	FALSE	05/01/2019	
4	Child	200	102	FALSE	05/01/2019	
5	Toy	50	300	FALSE	05/01/2019	
6						

```
15 }
16 } catch {
17   e → {
18     print[Str]('Error: ' + e);
19     0.0;
20   }
21 }
```

So that:

```
1 addBike('Electric',3000.0,20.0,false,'06/01/2019');
```

produces the following spreadsheet:



	A	B	C	D	E	F
1	Model	Price	Sales	Discontinued	Date	
2	Racer	1000.5	53	TRUE	05/01/2019	
3	Pro-Racer	2000.25	22	FALSE	05/01/2019	
4	Child	200	102	FALSE	05/01/2019	
5	Toy	50	300	FALSE	05/01/2019	
6	Electric	3000	20	FALSE	06/01/2019	
7						

Suppose that we want to calculate the total sales. This is done by the following function:

```
1 totalSales()::Float =
2   try {
3     let book::Excel = readWorkbook('esl/data/bikes.xlsx');
4     id(f::Float)::Float = f;
5     add(x::Float,y::Float)::Float = x + y; in
6     let s0::ExcelSheet = getSheet(book,0);
7     let sales::ExcelColumn = getColumnNamed(s0,'Sales'); in
8     let numbers = [ n | XFloat(n) ← sales ];
9     in foldr[Float,Float](id,add,0.0,numbers);
10  } catch {
11    e → {
12      print[Str]('Error: ' + e);
13      0.0;
14    }
15 }
```

Running `totalSales` produces:

```
1 total sales = 477.0
```

Suppose that we want to add a new entry into the spreadsheet. The following code adds a new row:

```
1 addBike(model::Str,price::Float,sales::Float,current::Bool,date
         ::Str)::Void =
2   try {
3     let book::Excel = readWorkbook('esl/data/bikes.xlsx');
4     newRow::ExcelRow = XRow([XStr(model),
5                               XFloat(price),
6                               XFloat(sales),
7                               XBool(current),
8                               XDate(date)]); in
9     let newBook::Excel =
10       case book {
11         XWorkbook([XSheet(name,rows)]) →
12           XWorkbook([XSheet(name,rows + [newRow])]); }
13     in {
14       writeWorkbook('esl/data/bikes.xlsx',newBook);
```

18. Java Interface

A Java class can be instantiated and used as an ESL actor. The class must implement the following interface:

```
public interface esl.lib.AlienActor {
  public ESLVal alienRef(String name);
  public void alienSend(ESLVal message);
  public void stop();
}
```

The class may be instantiated via the following:

```
1 new 'a.b.C'[T](x,y,z)
```

where `a.b.C` is the package path for the class `C`, `T` is a behaviour type describing the messages and names handled by the operation `alienSend` and `alienRef`. The arguments `(x,y,z)` are optional and are the ESL values supplied to the constructor for `C`.

EDB will display a Swing component when it is supplied to the operation: `edb.display`:

```
1edb.display::Forall[T] (Str,T) → Void
```

ans where the second argument is an instance of a class that implements the following interface:

```
public interface esl.lib.AlienActor {
  public ESLVal alienRef(String name);
  public void alienSend(ESLVal message);
  public void stop();
}
```

19. Compilation

This section will describe the compilation process and the Java constructs produced by the ESL compiler.

20. Implementation

This section will describe the Java libraries that support ESL.

21. Bibliography

The design of ESL was originally motivated by a need for a language to create simulation models for enterprises in order to support *what-if* analysis. The hypothesis for that work was that enterprises are complex and contain incomplete and unpredictable behaviour, therefore a bottom-up emergent approach such as that supported by actors and multi-agent systems is required. The design for a supporting technology was motivated by both existing actor languages, event-driven architecture and languages for enterprise architecture. A very early version of ESL was called LEAP which went through a number of iterations including a UML-like language supported by a modelling toolset.

Many of the ideas formulated to model enterprises are also applicable to modelling general complex hierarchical systems. By restricting the domain of interest to *mechanistic* systems we found that there was no lack of interesting research challenges, but that the problem of how to model human behaviour (intrinsic to enterprises) could be removed. Creating digital twins for complex mechanistic systems of systems requires language technology that supports emergent behaviour together with sense-making and adaptation arising from advanced approaches such as machine learning. This is the motivation for the current version of ESL described in this document. The bibliography given below documents many of the publications by the research team associated with ESL and serves to document the motivation for the current version.

- [1] Souvik Barat, Vinay Kulkarni, Tony Clark, and Balbir Barn. A simulation-based aid for organisational decision-making. In *Proceedings of the 11th International Joint Conference on Software Technologies (ICSOFT 2016) - Volume 2: ICSOFT-PT, Lisbon, Portugal, July 24 - 26, 2016*, pages 109–116, 2016.
- [2] Souvik Barat, Vinay Kulkarni, Tony Clark, and Balbir Barn. Enterprise modeling as a decision making aid: A systematic mapping study. In *The Practice of Enterprise Modeling - 9th IFIP WG 8.1. 2016*, pages 289–298, 2016.
- [3] Souvik Barat, Vinay Kulkarni, Tony Clark, and Balbir Barn. A method for effective use of enterprise modelling techniques in complex dynamic decision making. In *The Practice of Enterprise Modeling - 10th IFIP WG 8.1. 2017*, pages 319–330, 2017.
- [4] Souvik Barat, Vinay Kulkarni, Tony Clark, and Balbir Barn. A model based realisation of actor model to conceptualise an aid for complex dynamic decision-making. In *MODELSWARD 2017*, pages 605–616, 2017.
- [5] Souvik Barat, Vinay Kulkarni, Tony Clark, and Balbir Barn. A model based approach for complex dynamic decision-making. In *MODELSWARD 2017, Revised Selected Papers*, pages 94–118, 2017.
- [6] Souvik Barat, Vinay Kulkarni, Tony Clark, and Balbir Barn. An actor-model based bottom-up simulation - an experiment on indian demonetisation initiative. In *2017 Winter Simulation Conference, WSC 2017*, pages 860–871, 2017.
- [7] Balbir S. Barn, Tony Clark, and Vinay Kulkarni. Next generation enterprise modelling - the role of organizational theory and multi-agent systems. In *ICSOFT-EA 2014*, pages 482–487, 2014.
- [8] Balbir S. Barn, Tony Clark, and Vinay Kulkarni. Can organisational theory and multi-agent systems influence next generation enterprise modelling? In *Software Technologies - 9th International Joint Conference, ICSOFT 2014, Vienna, Austria, August 29-31, 2014, Revised Selected Papers*, pages 202–216, 2014.
- [9] Tony Clark. Introduction to the special issue on model-driven organisations. *Enterprise Modelling and Information Systems Architectures*, 13: 3:1–3:3, 2018. doi: 10.18417/emisa.13.3. URL <https://doi.org/10.18417/emisa.13.3>.
- [10] Tony Clark and Balbir Barn. Goal driven architecture development using LEAP. *Enterprise Modelling and Information Systems Architectures*, 8 (1):40–61, 2013. URL <https://www.emisa-journal.org/emisa/article/view/101>.
- [11] Tony Clark and Balbir S. Barn. Event driven architecture modelling and simulation. In *IEEE 6th International Symposium on Service Oriented System Engineering, SOSE 2011, Irvine, CA, USA, December 12-14, 2011*, pages 43–54, 2011.
- [12] Tony Clark and Balbir S. Barn. A common basis for modelling service-oriented and event-driven architecture. In *Proceeding of the 5th Annual India Software Engineering Conference, ISEC 2012, Kanpur, India, February 22-25, 2012*, pages 23–32, 2012.
- [13] Tony Clark and Balbir S. Barn. Dynamic reconfiguration of event driven architecture using reflection and higher-order functions. *Int. J. Software and Informatics*, 7(2):137–168, 2013. URL <http://www.ijsi.org/ch/reader/view\abstract.aspx?file_no=i157>.
- [14] Tony Clark and Balbir S. Barn. Using a projectional editor to implement interactive model navigation and view control. In *MODELSWARD 2015*, pages 338–345, 2015.
- [15] Tony Clark, Vinay Kulkarni, Balbir Barn, Robert B. France, Ulrich Frank, and Dan Turk. Towards the model driven organization. In *47th HICSS 2014*, pages 4817–4826, 2014.
- [16] Tony Clark, Dean Kramer, and Samia Oussena. Model driven reactive applications. *CoRR*, abs/1506.03380, 2015. URL <http://arxiv.org/abs/1506.03380>.
- [17] Tony Clark, Balbir Barn, Vinay Kulkarni, and Souvik Barat. Querying histories of organisation simulations. In *Information Systems Development 2017*, 2017.
- [18] Tony Clark, Vinay Kulkarni, Souvik Barat, and Balbir Barn. Actor monitors for adaptive behaviour. In *ISEC 2017*, pages 85–95, 2017.
- [19] Tony Clark, Vinay Kulkarni, Souvik Barat, and Balbir Barn. Generating filmstrip models from actor-based systems. In *Proceedings of MODELS 2017 Satellite Event*, pages 576–582, 2017.
- [20] Tony Clark, Vinay Kulkarni, Souvik Barat, and Balbir Barn. ESL: an actor-based platform for developing emergent behaviour organisation simulations. In *Advances in Practical Applications of Cyber-Physical Multi-Agent Systems 2017*, pages 311–315, 2017.
- [21] Tony Clark, Vinay Kulkarni, Balbir Barn, and Souvik Barat. The construction and interrogation of actor based simulation histories. In *ER Forum 2017*, pages 320–333, 2017.
- [22] Tony Clark, Vinay Kulkarni, Souvik Barat, and Balbir Barn. Actor-based methods, concepts and tools for analysing emergent behaviour: An introduction to a model based approach. In *Proceedings of the 11th Innovations in Software Engineering Conference, ISEC 2018*, pages 21:1–21:2, 2018.
- [23] Tony Clark, Vinay Kulkarni, Souvik Barat, and Balbir Barn. A homogeneous actor-based monitor language for adaptive behaviour. In *Programming with Actors - State-of-the-Art and Research Perspectives*, pages 216–244, 2018.
- [24] Rajesh K Karmani, Amin Shali, and Gul Agha. Actor frameworks for the jvm platform: a comparative analysis. In *Proceedings of the 7th International Conference on Principles and Practice of Programming in Java*, pages 11–20. ACM, 2009.
- [25] Vinay Kulkarni, Suman Roychoudhury, Sagar Sunkle, Tony Clark, and Balbir Barn. Modelling and enterprises - the past, the present and the future. In *MODELSWARD 2013*, pages 95–100, 2013.
- [26] Vinay Kulkarni, Tony Clark, Souvik Barat, and Balbir Barn. Model based enterprise simulation and analysis - A pragmatic approach reducing the burden on experts. In *Advances in Conceptual Modeling - ER 2014 Work-*

- shops*, pages 3–12, 2014.
- [27] Vinay Kulkarni, Souvik Barat, Tony Clark, and Balbir Barn. A wide-spectrum approach to modelling and analysis of organisation for machine-assisted decision-making. In *Enterprise and Organizational Modeling and Simulation - 11th International Workshop, EOMAS 2015, Held at CAiSE 2015, Stockholm, Sweden, June 8-9, 2015, Selected Papers*, pages 87–101, 2015.
 - [28] Vinay Kulkarni, Souvik Barat, Tony Clark, and Balbir Barn. Using simulation to address intrinsic complexity in multi-modelling of enterprises for decision making. In *SummerSim 2015*, pages 9:1–9:11, 2015.
 - [29] Vinay Kulkarni, Souvik Barat, Tony Clark, and Balbir S. Barn. Toward overcoming accidental complexity in organisational decision-making. In *MoDELS 2015*, pages 368–377, 2015.
 - [30] Joshua Chibuike Nwokeji, Tony Clark, Balbir S. Barn, and Vinay Kulkarni. A conceptual framework for enterprise agility. In *Proceedings of the 30th Annual ACM Symposium on Applied Computing, Salamanca, Spain, April 13-17, 2015*, pages 1242–1244, 2015.

A. ESL Syntax

Figures A.15 and A.16 define the syntax of ESL.

B. Displays

```

1 data Property = Prop(Str,Str);
2 type Props    = [Property];
3
4 propValue(props::Props,n::Str)::Str =
5   case props {
6     ps1 + [Prop(m,v)] + ps2 when m = n → v;
7   }
8
9 hasProp(props::Props,n::Str)::Bool =
10  case props {
11    ps1 + [Prop(m,v)] + ps2 when m = n → true;
12    props → false;
13  }
14
15 setProp(props::Props,n::Str,v::Str)::Props =
16  case props {
17    ps1 + [Prop(m,v0)] + ps2 when m = n →
18      ps1 + [Prop(n,v)] + ps2;
19  }
20
21
22 data PictureElement =
23   Rectangle(Int,Int,Int,Int,Str)
24 | Circle(Int,Int,Int,Str)
25 | Line(Int,Int,Int,Str)
26 | Image(Int,Int,Int,Str)
27 | Text(Int,Int,Str,Str);
28
29 data TreeElement =
30   TreeNode(PictureElement)
31 | VBox(Props,[TreeElement])
32 | HBox(Props,[TreeElement])
33 | Padding(Props)
34 | LabelledTree(Str,TreeElement);
35
36 data LinePoint = LPoint(Props,Int,Int);
37 data Row    = Row(Props,[Data]);
38 data Data   = Data(Props,Display);
39 data Slice  = Slice(Props,Str,Int);
40 data GLine  = GLine(Props,Str,[LinePoint]);
41 data Message = Message(Int,Int,Int,Str);
42 data Actor   = Actor(Int,Int,Int,Str);
43 data Node    = Node(Props,Int,Display);
44 data Edge    = Edge(Props,Int,Int,Display);
45
46 edgeSource(e::Edge)::Int = case e { Edge(ps,s,t,d) → s; }
47 edgeTarget(e::Edge)::Int = case e { Edge(ps,s,t,d) → t; }
48 nodeId(n::Node)::Int = case n { Node(ps,id,d) → id; }
49 nodeProps(n::Node)::Props = case n { Node(ps,id,d) → ps; }
50 setNodeProps(n::Node,ps::Props)::Node =
51  case n { Node(ps0,id,d) → Node(ps,id,d); }
52 edgeProps(e::Edge)::Props =
53  case e { Edge(ps,s,t,d) → ps; }
54 setEdgeProps(e::Edge,ps::Props)::Edge =
55  case e { Edge(ps0,s,t,d) → Edge(ps,s,t,d); }
56
57 data Display =
58   HTML(Str)
59 | Table(Props,[Row])
60 | Pie(Props,Str,Int,Int,[Slice])
61 | LineGraph(Props,Str,Str,Str,Int,Int,[GLine])
62 | Picture(Int,Int,[PictureElement])
63 | Sequence([Actor],[Message])

```

```

64 | Graph(Props,[Node],[Edge])
65 | Tree(Int,Int,TreeElement);

```

C. List Operations

```

1  adjoin[T](x::T,l::[T])::[T] =
2    if member[T](x,l)
3    then l;
4    else x:l;
5
6  select1[T](l::[T],d:T,p:(T)→Bool)::T =
7    case l {
8      [] → d;
9      h:t when p(h) → h;
10     h:t → select1[T](t,d,p);
11   }
12
13 map[M,N](f:(M)→N,l::[M])::[N] =
14  case l {
15    [] → [];
16    m:ms → (f(m)):map[M,N](f,ms);
17  }
18
19 remove[T](v::T,l::[T])::[T] =
20  case l {
21    h:t when (h=v) → remove[T](v,t);
22    h:t → h:remove[T](v,t);
23    [] → [];
24  }
25
26 remove1[T](v::T,l::[T])::[T] =
27  case l {
28    h:t when (h=v) → t;
29    h:t → h:remove1[T](v,t);
30    [] → [];
31  }
32
33 length[T](l::[T])::Int =
34  case l {
35    h:t → 1 + length[T](t);
36    [] → 0;
37  }
38
39 flatten[T](lists::[[T]])::[T] =
40  case lists {
41    h:t → h:flatten[T](t);
42    [] → [];
43  }
44
45 count[T](x:T,l::[T])::Int =
46  case l {
47    h:t → if h=x then 1+count[T](x,t); else count[T](x,t);
48    [] → 0;
49  }
50
51 hasPrefix[T](list::[T],prefix::[T])::Bool =
52  case list,prefix {
53    [],[] → true;
54    x:list,y:prefix when x=y → hasPrefix[T](list,prefix);
55    [],_ → false;
56  }
57
58 nth[T](l::[T],n::Int)::T =
59  case l {
60    h:t → if n = 0 then h; else nth[T](t,n-1);
61    [] → throw[T]('cannot take nth element.');
62  }
63
64 take[T](l::[T],n::Int)::[T] =
65  if n = 0
66  then []
67  else
68    case l {
69      h:t → h:(take[T](t,n-1));
70      [] → throw[T]('cannot take element ' + n);
71    }
72
73 drop[T](l::[T],n::Int)::[T] =
74  if n = 0
75  then l;
76  else
77    case l {
78      h:t → drop[T](t,n-1);
79      [] → throw[T]('cannot drop element ' + n);
80    }
81
82 subst[T](n::T,o::T,l::[T])::[T] =
83  case l {
84    [] → [];
85    h:t →
86      if h = o
87        then n:(subst[T](n,o,t));

```

1 module ::= exports imports binding*	A <i>module</i> is a unit of compilation. It can export any of the names it defines in its bindings. The exported names of each imported module are available.
2	
3	
4 exports ::= export (name Name) (, (name Name))* ;	Names of types, constructors, bindings can be exported. Reference a module as a path to the file.
5 imports ::= import str (, str)* ;	
6 binding ::= behBind actBind valBind typBind datBind;	
7 behBind ::= act name [gens] [args] typDec [extend] {behBody}	A behaviour may have generic type arguments, and may have initialisation arguments. The parent should be a behaviour.
8	
9 behBody ::= exports binding* [→ exp] arm*	
10 actBind ::= Act Name [extends typExp] {expDecls (messTyp ;)*}	A behaviour type may extend another behaviour type. The exported declarations and message handlers will be implemented by any behaviour with the type.
11	
12	
13 expDecls ::= [export dec (; dec)* ;]	A value binding may be generic. If the binding includes arguments then the value is a function.
14 valBind ::= name [gens] [args] typDec = command	
15	
16	
17 command ::= exp [;]	A command is an expression that is performed for its side-effect. A command that does not syntactically terminate with a ';' must be terminated by a '}'.
18	
19	
20 typBind ::= type Name [gens] = typExp [;]	A type definition may be generic and is terminated with a ';' when the expression is not terminated by a '}'.
21	
22	
23 datBind ::= data Name [gens] = termTyp (' ' termTyp)* ;	Note: Generic data types produce generic data constructors.
24 gens ::= '[' Name (, Name)* ']'	
25 args ::= '(' [dec (, dec)*] ')'	
26 dec ::= name typDec	
27 typExp ::= Int	<i>Integers.</i>
28 Float	Floats of the form Int '.' Int.
29 Str	Strings delimited by ' and '.
30 Bool	true or false.
31 Void	The type of a command - value is ignored.
32 [typExp]	Lists whose elements are of a given type.
33 ([typExp (, typExp)*) → typExp	Functions.
34 Bag {typExp}	Bags whose elements are of a given type.
35 Set {typExp}	Sets whose elements are of a given type.
36 Array [typExp]	Arrays whose elements are of a given type.
37 Hash [typExp,typExp]	Hash tables mapping keys to values.
38 Act { expDecls (messTyp ;)* }	A behaviour type.
39 union { [termTyp (, termTyp)*] }	A data type.
40 rec Name . typExp	A recursive type.
41 { dec (, dec)* }	A record type.
42 Forall [Name (, Name)*] typExp	A universal polymorphic type.
43 Name [typArgs]	Applying types to a generically defined type.
44 typArgs ::= '[' typExp (, typExp)* ']',	
45 termTyp ::= Name ['(' typExp (, typExp)* ')']	Matches supplied values against patterns, checks the optional guard, performs the command when the patterns match and the guard is satisfied.
46 typDec ::= :: typExp	
47 extend ::= extends exp	
48 arm ::= pat (, pattern)* [when exp] → command	Patterns with optional ':' (cons) and '+' (append).
49	
50	
51 pattern ::= atomPat [: pat + pat]	Variable.
52 atomPat ::= name	Integer constant pattern.
53 int	Term pattern (with optional generic types).
54 Name [typArgs] [patArgs]	List pattern.
55 '[' [exp (, exp)*] ']',	String constant pattern.
56 str	Constant boolean pattern.
57 true	Constant boolean pattern.
58 false	Non-empty bag pattern.
59 bag{pattern pattern}	Empty bag pattern.
60 bag{}	Non-empty set pattern.
61 set{pattern pattern}	Empty set pattern.
62 set{}	
63 '(' pattern ')'	
64 patArgs ::= '(' pattern (, pattern)* ')'	

Figure A.15: ESL Syntax (1 of 2)

1	exp ::= [exp (, exp)*]	
2	exp ::= if exp then exp else exp	Conditional expression. Type of then-exp and else-exp should agree.
3	fun[exp](args)dec exp	The optional expression is used as the name of the function.
4	grab(exps) exp	The locks are acquired simultaneously.
5	probably '(' exp ')', exp [else exp]	Probability based on a value out of 100.
6	let binding* in exp	No names can reference each other.
7	plet binding* in exp	Parallel binding, waits until all value expressions have completed.
8	letrec binding* in exp	Mutually recursive binding for functions. Values established in sequence.
9	case exps '{', arm* '}'	Number of values must agree with number of patterns in arms.
10	for pattern in exp do exp	If a value does not match pattern then it is ignored.
11	find pattern in exp do exp else exp	Selection from a list.
12	try exp catch '{', arm* '}'	Each arm should have a single pattern that matches a thrown value.
13	throw '[' typExp ']', exp	Throw to the most recently established try. The type is required for the checker.
14	bag '{', exps [' ', exp] '}'	Add elements to an existing bag, or the empty bag
15	set '{', exps [' ', exp] '}'	Add elements to an existing set, or the empty set.
16	'{', name = exp (; name = exp)* '}'	A record.
17	'{', command* '}'	Perform commands in sequence. Return value of final command.
18	new str '[' typExp ']', ['(' exps ')',]	Create an instance of a Java class that implements AlienActor.
19	new Array '[' typExp ']', ['(' exp ')',]	Create an array of given type and length.
20	new Hash '[' typExp , typExp ']',	Create a hash table with supplied key types and value types.
21	new exp ['(' exps ')',]	Create an actor with the given behaviour.
22	become exp ['(' exps ')',]	Change behaviour (the new behaviour must be of the same type as self).
23	exp op exp	Binary expression.
24	not exp	Boolean negation.
25	exp ← Name ['(' exps ')',]	Send a message.
26	exp [typArgs] ['(' exps ')',]	Apply a (possibly generic) function.
27	exp '[' exp ']',	Reference an array element by index.
28	exp . name	Reference a record field or an exported name of an actor.
29	- exp	Negation.
30	int	Integer constant.
31	float	Float constant.
32	str	String constant.
33	char	Character constant (actually an integer).
34	true	Boolean constant.
35	false	Boolean constant.
36	self	The currently executing actor (constant) or null if no actor.
37	super	The parent actor.
38	now	Current time in milliseconds since application started to run.
39	null	Null is polymorphically an instance of all types.
40	name	A program identifier starting with a lower case letter.
41	'[' exps ']',	A list.
42	'[' exp ']' qual (, qual) ']',	A list comprehension.
43	Name [typArgs] ['(' exps ')',]	A term. Type args required if the corresponding data definition was generic.
44	'(' exp ')',	
45	qual ::= pattern ← exp	Iteration through a list. Elements are ignored if they do not match the pattern.
46	? exp	A guard expression in a comprehension.
47	op ::= @	Concurrent binary operator. Returns whichever operand completes first.
48	or and	Boolean operators.
49	= <> < <= > >=	Comparison operators.
50	* / + - %	Arithmetic operators.
51	char ::= #c	Character c represented as ASCII code.
52	#\space	ASCII code for space character.
53	#\newline	ASCII code for newline character.
54	#\backslash	ASCII code for backslash character.

Figure A.16: ESL Syntax (2 of 2)

```

88     else h:(subst[T](n,o,t));
89   }
90
91 head::Forall[T]([T])→T = fun(l:[T]):[T]
92 case 1 {
93   h:t → h;
94   [] → throw[T]('cannot take the head of []');
95 }
96
97 tail::Forall[T]([T])→T = fun(l:[T]):[T]
98 case 1 {
99   h:t → t;
100  [] → throw[T]('cannot take the tail of []');
101 }
102
103 isNil[T](l:[T]):Bool =
104 case 1 {
105   [] → true;
106   l → false;
107 }
108
109 member[T](e:T,l:[T]):Bool =
110 case 1 {
111   [] → false;
112   x:xs when x = e → true;
113   x:xs → member[T](e, xs);
114 }
115
116 reverse[T](l:[T]):[T] =
117 case 1 {
118   [] → [];
119   x:xs → reverse[T](xs) + [x];
120 }
121
122 exists[T](pred:(T)→Bool,l:[T]):Bool =
123 case 1 {
124   [] → false;
125   x:xs when pred(x) → true;
126   x:xs → exists[T](pred, xs);
127 }
128
129 forall::[T](pred:(T)→Bool,l:[T]):Bool =
130 case 1 {
131   [] → true;
132   x:xs when pred(x) → forall[T](pred, xs);
133   x:xs → false;
134 }
135
136 replaceNth[T](l:[T],n:Int,x:T):[T] =
137 case 1 {
138   [] → throw[[T]]('cannot replace nth of []');
139   h:t when n=0 → x:t;
140   h:t → h:replaceNth[T](t,n-1,x);
141 }
142
143 indexOf[T](t:T,l:[T]):Int =
144 case 1 {
145   [] → -1;
146   h:l when h=t → 0;
147   h:l → 1 + indexOf[T](t,l);
148 }
149
150 select[T](p:(T)→Bool,l:[T]):[T] =
151 case 1 {
152   [] → [];
153   h:t when p(h) → h:select[T](p,t);
154   h:t → select[T](p,t);
155 }
156
157 reject[T](p:(T)→Bool,l:[T]):[T] =
158 case 1 {
159   [] → [] [T];
160   h:t when p(h) → reject[T](p,t);
161   h:t → h:reject[T](p,t);
162 }
163
164 last[T](l:[T]):T =
165 case 1 {
166   [] → throw [T]('no last element of empty list');
167   x:[] → x;
168   h:l → last[T](l);
169 }
170
171 butlast[T](l:[T]):[T] =
172 case 1 {
173   [] → [];
174   [x] → [];
175   h:l → h:butlast[T](l);
176 }
177
178 occurrences[T](x:T,l:[T]):Int =
179 case 1 {
180   [] [T] → 0;
181   h:t when h=x → 1 + occurrences[T](x,t);
182   h:t → occurrences[T](x,t);
183 }
184
185 filter[T](pred:(T)→Bool,l:[T]):[T] =
186 case 1 {
187   [] → [];
188   h:t →
189     if pred(h)
190       then h:filter[T](pred,t);
191     else filter[T](pred,t);
192 }
193
194 foldr[M,N](map:(M)→N,op:(N,N)→N,empty:N,list::[M]):N =
195 case list {
196   [] → empty;
197   h:t → op(map(h),foldr[M,N](map,op,empty,t));
198 }
199
200 id[T]:([T]→T) = fun(x:T):T x;
201
202 sum(list::[Int]):Int =
203 foldr[Int,Int](id[Int],fun(x:Int,y:Int):Int x+y,0,list);
204
205 removeDups[T](list::[T]):[T] =
206 case list {
207   [] → [];
208   x:l → x:removeDups[T](remove[T](x,l));
209 }
210
211 prefixes[T](l:[T]):[[T]] =
212 letrec pre(prefixes:[[T]],rest:[T]):[[T]] =
213   if rest = []
214     then prefixes;
215   else pre(prefixes + [
216     last[[T]](prefixes)+[head[T](rest)]
217   ],tail[T](rest));
218   in pre([],l);
219
220 takeWhile[T](p:(T)→Bool,l:[T]):[T] =
221 case 1 {
222   [] → [];
223   t:l when p(t) → t:takeWhile[T](p,l);
224   t:l → [];
225 }
226
227 dropWhile[T](p:(T)→Bool,l:[T]):[T] =
228 case 1 {
229   [] → [];
230   t:l when p(t) → dropWhile[T](p,l);
231   l → l;
232 }
233
234 removeAll[T](removables:[T],l:[T]):[T] =
235 case removables {
236   [] → l;
237   t:r → removeAll[T](r,remove[T](t,l));
238 }

```